/* прим. пер.: все примеры в этой книге при создании и сохранении модулей используют нотацию типа C:Start. Начиная с версии CA-Visual Objects 2.5b (см. файл Watsnew.doc) в именах модулей более не допускается использование символов \ / : * ? " < > | .*/

Класс PRINTER - Золушка CA-Visual Objects

Печатный отчет является, без сомнения, конечным результатом труда пользователей ваших программ и внешний вид отчета весьма важен для них. Класс Printer предоставляет то же визуальное качество отчета, что и другие приложения на основе ОС Windows, такие, как текстовые процессоры. Этот класс имеет определенные преимущества перед генератором отчетов, имеющимся в состава CA-Visual Objects и продуктами сторонних производителей. Мы назвали этот класс "Золушкой", потому что многие разработчики не реализуют его потенциальные возможности.

Два основных преимущества этого класса перед встроенным генератором отчетов заключаются в том, что, во-первых, он не использует ODBC, и, соответственно, нет необходимости в настройке ODBC, а во-вторых,  он не требует использования большого числа динамических библиотек DLL, нуждающихся в поддержке и сопровождении. Причина этого кроется в том, что класс Printer наследуется от класса Window и его код содержится в той же библиотеке GUI.DLL, что и  код других окон. Некоторые методы класса Window используются для отображения графических или текстовых объектов на экране. Можно сказать, что приведенный ниже код класса Printer будет работать более эффективно, чем код, созданный каким бы то ни было визуальным редактором. Это может оказаться весьма уместным для больших отчетов или загруженной сети.

Недостатками класса являются отсутствие визуального редактора для разработки форм отчетов, отсутствие режима просмотра, поддержки формата RTF и выравнивания текста по обоим краям. В то же время, есть такие раскладки страниц отчета, которые удобнее делать именно с помощью класса Printer. Предварительный просмотр отчета, конечно же, проигрывает просмотру в текстовом процессоре, поскольку пользователь лишен возможности изменять раскладку. Просмотр данных может быть осуществлен через окно данных (DataWindow) и т.п. без просмотра раскладки страницы. Некоторые из упомянутых ограничений безусловно преодолимы, но не в этой жизни (пардон, главе…)

Далее вы увидите, что мы сможем:

( печатать, используя любой из шрифтов, имеющихся на машине;

( использовать нескольких шрифтов в одном и том же документе;

( устанавливать поля;

( определять колонки, заголовки, подножия, тела отчетов, суммирование;

( выравнивать текст;

( использовать графику;

( печатать тексты и мемо-файлы;

( управлять переносом слов;

( печатать многоколоночные отчета типа телефонных списков и т.п.

В этом безусловно проявляется гибкость класса Printer.

Тем не менее, нам требуется провести некоторую подготовительную работу, чтобы получить максимальную отдачу. Как мы увидим в дальнейшем, сам по себе, в "чистом" виде, класс Printer может не очень много.

Это очень длинная глава, поскольку представлено много кода. Одним из недостатков класса является необходимость разработки дополнительного кода, но значительная часть этого кода представлена в виде библиотеки. Это хорошая возможность увидеть принцип наследования в работе. Предполагается, что библиотека, разработанная ниже (ReportPrinter Library), может быть непосредственно использована вами в работе со своими приложениями, и, конечно же, вы можете улучшить ее.

Прежде, чем приступить к проверке примеров, следует удостовериться, что принтер подключен к компьютеру и установлен как принтер по умолчанию. Читатели, еще недавно работавшие в ДОС убедятся, что способ управления различными принтерами, использованный в ОС Windows, является одной из наиболее привлекательных черт этой ОС. Больше не нужны написанные вручную драйверы принтеров с ESC-последовательностями.

BasicPrinter Class - базовый класс Printer

Мы начнем с разработки приложения с основными возможностями печати. В проекте EVO создайте новое приложение и назовите его "Ch15". Отметьте опцию Debug, отключите опцию ReadMe, а в состав приложения включите библиотеки System Library и GUI Library. Нажмите ОК. Переименуйте вновь созданный Module1 в "C:Start", откройте редактор кода и введите следующий код:

METHOD Start() CLASS App

LOCAL oPrinter AS BasicPrinter

oPrinter := BasicPrinter{"Basic Test"}

oPrinter:Start()

oPrinter:Destroy()

RETURN NIL

CLASS BasicPrinter INHERIT Printer

METHOD PrinterExpose(oPEE) CLASS BasicPrinter

SELF:TextPrint("Print test at 0,0", Point{0,0})

SELF:TextPrint("Print test at 10,10", Point{10,10})

SELF:TextPrint("Print test at 100,100", Point{100,100})

SELF:TextPrint("Print test at 1000,1000", Point{1000,1000})

SUPER:PrinterExpose(oPEE)

RETURN FALSE
Сохраните код, закройте редактор, скомпилируйте и запустите приложение.

В приведенном выше коде нами создан класс BasicPrinter, наследуемый от класса Printer. Объект этого класса создан в методе App:Start() с параметром "Basic Test". Этот параметр называется JobName - Имя Задачи. Это имя можно увидеть в окне менеджера печати Windows. Метод Printer:Start() вызывает метод Printer:PrinterExpose() до тех пор, пока он не возвратит логическое выражение FALSE. В упомянутом коде метод PrinterExpose() возвращает жестко закодированное значение FALSE, поэтому будет напечатана только одна страница. После этого объект PrinterExpose() уничтожается и приложение заканчивает свою работу.

Для тех читателей, которые знакомы с классом Printer в ранних версиях CA-Visual Objects, сообщаю, что больше нет необходимости проверять статус принтера с помощью метода Printer:IsValid().

Вывод страницы зависит от кода в методе (или вызываемого из метода) BasicPrinter:PrinterExpose(). Каждая из четырех строк создается через обращение к Windows:TextPrint() с указанием печатаемого текста и его расположения на странице, передаваемых в качестве параметров. (Вспомните, BasicPrinter унаследован от Printer, который, в свою очередь, наследуется от  Window). Размещение текста на странице зависит от разрешения принтера, которое обычно находится в пределах от 60 до 1200 пиксел на дюйм. При разрешении в 60 пиксел последняя строка, с координатами 1000, 1000 окажется за пределами страницы. Хотя это и не вызовет ошибки, однако является явно бесполезным.

Мы разработаем код, который позволит определять местоположение на странице в привычных х-у координатах в диапазоне от 0 до 100, независимо от размеров страницы и количества пиксел.

Для простоты мы усовершенствуем приложение таким образом, чтобы часть данных печаталась на одной странице, а часть - на другой, при этом на каждой странице должен быть ее номер. Для этого изменим класс BasicPrinter.

CLASS BasicPrinter INHERIT Printer

PROTECT dwPage AS DWORD

METHOD PrinterExpose(oPEE) CLASS BasicPrinter

SELF:dwPage += 1

IF SELF:dwPage = 1

SELF:TextPrint("Print test at 300, 300", Point{300, 300})

SELF:TextPrint("Print test at 400, 400", Point{400, 400})

ELSE

SELF:TextPrint("Print test at 500, 500",Point{500, 500})

SELF:TextPrint("Print test at 1000, 1000", Point{1000, 1000})

ENDIF

SELF:TextPrint("Страница No: " + NTrim(SELF:dwPage), Point{0, 0})

SUPER:PrinterExpose(oPEE)

RETURN !(SELF:dwPage >= 2)
Сохраните код, закройте редактор, скомпилируйте и запустите приложение.

Теперь dwPage является свойством класса BasicPrinter и значение переменной увеличивается внутри метода PrinterExpose(). С помощью этого свойства выводится номер страницы, а другие печатаемые элементы изменили свое местоположение. Печать осуществляется теперь на двух страницах, поскольку, подчеркнем это особо, метод Start() последовательно вызывает метод PrinterExpose() до тер пор, пока последний не возвратит FALSE, а этого не случится, пока SELF:dwPage не примет значение равное 2.

Давайте еще модифицируем код, чтобы обеспечить возможность печати некоторого диапазона страниц. Может показаться, что коль скоро объект Range передан в метод Printer:Start(), то объект Printer напечатает именно этот диапазон страниц, игнорируя страницы, не попадающие в диапазон. Однако, это не так. Вероятно, это происходит потому, что мы не можем уничтожить объекты, уже обработанные методом PrinterExpose(). Особенно важно то, что установка диапазона определяет количество печатаемых страниц, но не определяет начальную страницу. К тому же, диапазон никак не связан с конкретным количеством страниц SELF:dwPage, если только мы сами не свяжем их. Итак, мы добавим свойство Range в класс BasicPrinter, однако переместим момент начала печати в наш собственный метод BasicPrinter:Start(). Замените введенный ранее вами код в редакторе на приведенный ниже. Далее мы объясним его.

METHOD Start() CLASS App

LOCAL oPrinter AS BasicPrinter

oPrinter := BasicPrinter{"Basic Test"}

oPrinter:Range := Range{2, 2}

oPrinter:Start()

oPrinter:Destroy()

RETURN NIL

CLASS BasicPrinter INHERIT Printer

PROTECT dwPage AS DWORD

EXPORT Range AS OBJECT

PROTECT lNoPrint AS LOGIC

METHOD Init(cJobname) CLASS BasicPrinter

SELF:dwPage := 1

SUPER:Init(cJobName)

RETURN SELF

METHOD Start() CLASS BasicPrinter

IF IsInstanceOf(SELF:Range, #Range)

SELF:lNoPrint := TRUE

DO WHILE SELF:dwPage < SELF:Range:Min

SELF:ReportExpose()

SELF:dwPage += 1

ENDDO

SELF:lNoPrint := FALSE

SUPER:Start(SELF:Range)

SELF:Range := NULL_OBJECT

ELSE

SELF:lNoPrint := FALSE

SUPER:Start()

ENDIF

RETURN NIL

METHOD ReportExpose() CLASS BasicPrinter

IF SELF:dwPage = 1

SELF:PrintText("Print test at 300, 300", Point{300, 300})

ELSE

SELF:PrintText("Print test at 300, 300 "+ Str3(SELF:dwPage * 10, 3, 0), Point{500, 500})

ENDIF

SELF:PrintText("Страница No: " + NTrim(SELF:dwPage), Point{0, 0})

RETURN !(SELF:dwPage > 4)

METHOD PrinterExpose(oPEE) CLASS BasicPrinter

LOCAL lRet AS LOGIC

lRet := SELF:ReportExpose()

SELF:dwPage += 1

SUPER:PrinterExpose(oPEE)

RETURN lRet

METHOD PrintText(cString, oPoint) CLASS BasicPrinter

IF ! SELF:lNoPrint

SELF:TextPrint(cString, oPoint)

ENDIF

RETURN NIL
Обратите внимание на то, что свойство Range типизировано как Object, а не как Range для получения максимальной отдачи от метода BasicPrinter:Start(). По окончании его использования можно присвоить ему нулевое значение - NULL_OBJECT.

Строка 4 метода App:Start() устанавливает диапазон от 2 (мин.) до 2 (макс.) Иными словами, будет напечатана только вторая страница. Если закомментировать эту строку, то будут напечатаны все 5 страниц диапазона. Однако, зачем нужны методы ReportExpose()  и PrinterExpose()?

Дело в том, что не существует непосредственной связи между данными и номерами страниц. Зачастую единственным способом узнать, что именно следует печатать на первой странице заданного диапазона, является разложить весь отчет на страницы, но не печатать его. Метод PrintText() лишь вызывает метод TextPrint(), который реально ничего не печатает, если свойство lNoPrint возвращает FALSE. Метод ReportExpose() определяет конец страницы, независимо от того, выводится она на печать или нет. В пределах заданного диапазона, или если диапазон вообще не был указан, метод PrinterExpose() работает также, как и раньше, только делает это через вызов ReportExpose().

Основным различием является то, что теперь раскладка обрабатывается методом ReportExpose(), а фактический вывод на печать зависит от логического значения. Это будет использовано нами ниже при создании более сложных отчетов, где мы будем наследовать от эквивалента класса BasicPrinter - класса ReportPrinter. Наследуемые классы будут иметь метод ReportExpose() и не будут иметь своих методов Start() и PrinterExpose().

Сохраните код, закройте редактор, скомпилируйте и запустите приложение, чтобы убедиться, что будет напечатана только вторая страница. Затем закомментируйте строку 4 метода App:Start(). Вновь скомпилируйте и запустите приложение. Теперь должны напечататься все 5 страниц.

С учетом этой информации мы можем напечатать текст на любом количестве страниц и в любом нужном месте страницы. Мы обеспечили возврат FALSE при окончании данных, а также возможность задания диапазона печати. Однако, теперь нам нужно обеспечить независимость написанного выше кода от типа и разрешения используемого принтера.

Класс PrintingDevice

Очень важно понять разницу между классами Printer и PrinterDevice. Последний содержит информацию, относящуюся к аппаратной части принтера и такие текущие установки как ориентацию страницы, размер бумаги (А4 и т.п.), число копий и т.д. Мы сами часто называем устройства печати принтерами, что влечет за собой некоторую путаницу.

Класс Printer содержит информацию о том, что должно быть напечатано и где именно - в каком месте страницы. Одновременно одна из переменных ссылается на соответствующее объект устройства PrintingDevice, давая таким образом доступ классу Printer к размеру бумаги и т.п.

В приведенных примерах класс PrintingDevice даже не упомянут. В каждом случае обращения к BasicPrinter предполагалось, что требуемым устройством печати является принтер, установленный в Windows по умолчанию. Мы могли бы передать объект PrintingDevice в качестве второго параметра в метод BasicPrinter:Init(). В этом случае мы могли бы устанавливать альбомную ориентацию листа бумаги перед передачей в класс Printer. Более подробно мы обсудим это позднее.

Будьте организованны!

Мы должны иметь возможность расположить текст на странице независимо от разрешения принтера и размера листа бумаги, находящегося в принтере. Возможность задавать координаты вывода в виде процентов от ширины и высоты размера листа сделала бы жизнь намного легче.

Большинство принтеров имеют непечатаемые участки - поля - по всему периметру страницы. Конкретные значения зависят от модели принтера. Мы решим эту проблему путем задания своих собственных полей с учетом имеющихся аппаратных ограничений так, чтобы если аппаратные поля больше установленных нами, то мы могли бы уменьшить наши поля, и наоборот. 

Предположим, что существует некий файл INI, содержащий следующую секцию:

[HP LaserJet 6L]

Left=2

Right=2

Top=2

Bottom=2
Эти значения могут быть изменены пользователем, например, для изменения размещения текста на странице, или добавить секцию для другого принтера. Указанные значения представляют собой проценты от ширины и высоты аппаратно определенной зоны печати на листе бумаги - в данном случае, все поля равны 2%. Они не обязательно должны быть целочисленными, но в действительности имеют значение только незначительные изменения. Установка значений, скажем, в 20%, будет непрактичной, поскольку изменится положение только текста и графики, размер же области печати останется неизменным. Тем не менее, как мы увидим, в  подобном подходе тоже есть зерно истины, во всяком случае, реализовать его  значительно легче, нежели изменять код приложения. Имя секции соответствует имени печатающего устройства (Printing Device), в данном случае - HP Laser Jet 6L. Ниже мы увидим, как можно использовать это. Различные принтеры могут иметь отличные от приведенных значения полей.

Говоря "Будьте организованны", в конце главы мы увидим, что нет предела совершенству, но тем не менее, для начала уже неплохо.

ReportPrinter - класс Printer высокого уровня

Мы создадим класс Printer, назовем его ReportPrinter, который наследуется от стандартного класса Printer, но он будет унаследован  несколько необычном способом. Поскольку ReportPrinter потенциально может быть использован в различных приложениях, мы поместим его в специальную библиотеку ReportPrinter Library. Эта библиотека будет содержать также и другие необходимые классы.

Для начала напомним, как используются библиотеки. Библиотека создается в интегрированной среде разработки как если бы она была обычным приложением. После создания ее код может быть экспортирован в файл AEF для резервного копирования или переноса на другой компьютер. Библиотеки компилируются также, как и приложения. Множество приложений может обращаться к этой библиотеке также, как осуществляются обращения к стандартным библиотекам CA-Visual Objects. Компилированный код библиотеки встраивается в исполнительный файл, поэтому библиотеки имеют самостоятельное представление только на компьютере разработчика, но не на компьютере пользователя. 

Для тестирования библиотеки нам необходимо приложение, файлы данных и классы DBServer. Файлы данных используются те же, что и раньше, использованные при создании приложения Video, поэтому они должны присутствовать на вашем ПК. Для удобства классы DBServer представлены в виде библиотеки в файле AEF в каталоге Ch15 на сопроводительном CD, поскольку от ручного ввода всего этого кода нет ни малейшей пользы. Библиотека ReportPrinter Library также представлена в виде AEF как приложение ReportPrinter, однако мы настоятельно рекомендуем вам начать создание библиотеки и приложения так, как описано ниже. Файлы AEF содержат дополнительный код по сравнению с приведенным ниже. В тот момент, когда потребуется дополнительный код, мы порекомендуем вам импортировать его с CD. Поэтому не тратьте время попусту.

Когда приложение обращается к библиотеке, последняя должна находится в том же Проекте и должна быть уже скомпилирована. В противном случае создание (компиляция) кода приложения породит множество ошибок. Создавая библиотеку, следует компилировать ее код также, как и любое приложение для проверки наличия ошибок. Однако, ее работоспособность вы сможете проверить только при наличии приложения, использующего ее.

Если вы решите не создавать библиотеку описанным ниже способом, вам будет необходимо сначала импортировать и скомпилировать два библиотечных экспортных файла, а затем - импортировать и скомпилировать само приложение. Вот эти три файла AEF:

( ReportPrinter Library,

( ReportPrinter Server Library,

( ReportPrinter App.

Если же вы последуете за нами в процессе создания библиотек, в нужно время вам подскажут, когда нужно импортировать указанные выше файлы.

Создание библиотеки Ch15
Если вы до сих пор находитесь в редакторе кода, закройте его и возвратитесь в репозиторий.

Мы создадим приложение ReportPrinter, со всеми библиотеками, однако используем при этом такие имена как Ch15 App, Ch15 Library и Ch15 Servers, чтобы впоследствии можно было импортировать полные их версии с CD, сохранив наш собственный код.

Создайте новую пустую библиотеку со вкладки Basic Галереи приложений в проекте EVO и назовите ее Ch15 Library. Измените название модуля, созданного по умолчанию, на "C:ReportWriter". Откройте свойства приложения и на вкладке Libraries включите в ее состав System Classes, GUI Classes и RDD Classes. Щелкните ОК. Обратите внимание на значок приложения. Он больше похож на папку с файлами.

Откройте редактор кода и импортируйте файл Code1.prg из каталога Ch15 компакт-диска. Вы получите следующий код:

CLASS ReportPrinter INHERIT Printer

PROTECT Owner AS Window

PROTECT dwPage AS DWORD

PROTECT dwPageWidth AS DWORD

PROTECT dwPageHeight AS DWORD

PROTECT oPrintArea AS BoundingBox

PROTECT NoPrint AS LOGIC

EXPORT Range AS OBJECT

DECLARE METHOD PrintGrid
METHOD Start() CLASS ReportPrinter

LOCAL lMorePages AS LOGIC

IF IsInstanceOf(SELF:Range, #Range)

SELF:NoPrint := TRUE

DO WHILE SELF:dwPage < SELF:Range:Min

IF IsMethod(SELF, #ReportExpose)

lMorePages := Send(SELF, #ReportExpose)

SELF:dwPage += 1

IF ! lMorePages

EXIT

ENDIF

ELSE

EXIT

ENDIF

ENDDO

IF lMorePages

SELF:NoPrint := FALSE

SUPER:Start(SELF:Range)

ELSE

Textbox{SELF:Owner, "Печать", "Нечего печатать", ;

BOXICONEXCLAMATION}:Show()

ENDIF

SELF:Range := NULL_OBJECT

ELSE

SELF:NoPrint := FALSE

SUPER:Start()

ENDIF

RETURN NIL
METHOD PrinterExpose(oPEE) CLASS ReportPrinter

LOCAL lRet AS LOGIC

IF IsMethod(SELF, #ReportExpose)

lRet := Send(SELF, #ReportExpose)

ENDIF
SELF:dwPage += 1

SUPER:PrinterExpose(oPEE)

ApplicationExec(EXECWHILEEVENT)

RETURN lRet
METHOD Get_Margin(cEdge, lWidth) CLASS ReportPrinter

LOCAL r4Ret AS REAL4

r4Ret := oINI:GetInt(oPrintDev:Device, cEdge)

IF lWidth

r4Ret *= SELF:dwPageWidth

ELSE

r4Ret *= SELF:dwPageHeight

ENDIF

RETURN r4Ret / 100
METHOD Init(Owner, cJobName) CLASS ReportPrinter

LOCAL r4LeftMargin, r4RightMargin, r4TopMargin, r4BottomMargin, r4PrintWidth, r4PrintHeight AS REAL4

LOCAL o_Origin AS Point

LOCAL oDimension AS Dimension

IF IsNil(Owner)

Textbox{ , "Класс ReportPrinter", "Владелец == NIL"}:Show()

BREAK

ELSE

IF ! IsInstanceOf(Owner, #Window)

Textbox{ , "Класс ReportPrinter ", "Владелец неизвестен."}:Show()

BREAK

ENDIF

ENDIF
Default(@cJobName, "VO Print Job")

IF ! IsInstanceOf(oPrintDev, #PrintingDevice)

oPrintDev := PrintingDevice{}

ENDIF
SUPER:Init(cJobName, oPrintDev)

SELF:Owner := Owner

SELF:Idle()

SELF:dwPageWidth := SELF:CanvasArea:Width

SELF:dwPageHeight := SELF:CanvasArea:Height

IF oPrintDev:Orientation = DMORIENT_PORTRAIT

r4LeftMargin := SELF:Get_Margin("Left", TRUE)

r4RightMargin := SELF:Get_Margin("Right", TRUE)

r4TopMargin := SELF:Get_Margin("Top", FALSE)

r4BottomMargin := SELF:Get_Margin("Bottom", FALSE)

r4PrintWidth := SELF:dwPageWidth - r4LeftMargin - r4RightMargin

r4PrintHeight := SELF:dwPageHeight - r4BottomMargin - r4TopMargin

o_Origin := Point{r4LeftMargin, r4BottomMargin}

oDimension := Dimension{r4Printwidth, r4PrintHeight}

ELSE

r4LeftMargin := SELF:Get_Margin("Left", FALSE)

r4RightMargin := SELF:Get_Margin("Right", FALSE)

r4TopMargin := SELF:Get_Margin("Top", TRUE)

r4BottomMargin := SELF:Get_Margin("Bottom", TRUE)

r4PrintWidth := SELF:dwPageWidth - r4TopMargin - r4BottomMargin

r4PrintHeight := SELF:dwPageHeight - r4LeftMargin - r4RightMargin

o_Origin := Point{r4BottomMargin, r4RightMargin}

oDimension := Dimension{r4Printwidth, r4PrintHeight}

ENDIF
SELF:oPrintArea := BoundingBox{o_Origin, oDimension}

o_Origin:Destroy()

oDimension:Destroy()

SELF:Font := Font{FONTSWISS10}

CollectForced()

RETURN SELF

ACCESS dwPage() CLASS ReportPrinter

RETURN SELF:dwPage

GLOBAL oPrintDev as PrintingDevice

GLOBAL oINI as INIFileSpec

METHOD Destroy() CLASS ReportPrinter

SELF:oPrintArea := NULL_OBJECT

CollectForced()

SUPER:Destroy()

RETURN NIL

METHOD PrintFrame(dwPen) CLASS ReportPrinter

LOCAL aFrame AS ARRAY

LOCAL oPoint AS Point

LOCAL oDim AS Dimension

LOCAL oPen AS Pen

LOCAL oLine AS LineObject

LOCAL dwCounter AS DWORD

IF ! SELF:NoPrint

aFrame := {{0, 0}, {0, SELF:dwPageHeight - 1}, {SELF:dwPageWidth - 1, SELF:dwPageHeight - 1}, ;

{SELF:dwPageWidth - 1, 0}, {0, 0}}

oPoint := Point{0, 0}

oDim := Dimension{0, 0}

oPen := Pen{ , , dwPen}

oLine := LineObject{oPoint, oPoint, oPen}

FOR dwCounter := 1 UPTO 4

oPoint:X := aFrame[dwCounter, 1]

oPoint:Y := aFrame[dwCounter, 2]

oDim:Height := aFrame[dwCounter + 1, 2] - aFrame[dwCounter, 2]

oDim:Width := aFrame[dwCounter + 1, 1] - aFrame[dwCounter, 1]

oLine:Origin := oPoint

oLine:Size := oDim

oLine:Pen := oPen

SELF:Draw(oLine)

NEXT

oPoint := NULL_OBJECT

oDim := NULL_OBJECT

oPen := NULL_OBJECT

oLine := NULL_OBJECT

CollectForced()

ENDIF

RETURN NIL

Прежде всего обратите внимание на 2 глобальные переменные. Они объявлены в библиотеке, поскольку все приложения - пользователи библиотеки будут нуждаться в них, однако конкретные значения им будут присвоены в методах Start() использующих их приложений. <oPrintDev> может быть установлен следующим образом:

oPrintDev := PrintingDevice{}
С помощью этой строки будет обнаружен принтер, установленный по умолчанию в ОС Windows, загружена информация о нем и она станет доступной для наших классов по мере их создания. Файл помощи CA-Visual Objects объясняет, как использовать другие устройства печати, если известны их аппаратные параметры. Мы будем использовать принтер по умолчанию, хотя метод PrintingDevice:Setup() позволяет выбрать и другой принтер в сети.

Метод Start() также позволяет связать номер страницы с диапазоном, как мы сделали это для класса BasicPrinter, но чуть сложнее, поскольку мы предварительно проверяем наличие метода ReportExpose(). Без этого библиотека не может быть скомпилирована без выдачи предупреждений компилятором. Можно включить пустой метод ReportExpose() во избежание появления таких предупреждений. Он также выводит окно с предупреждением, если пользователь пытается задать диапазон печати, выходящий за конец отчета.

Доступ к файлу INI, упомянутому ранее в Get_Margin(), обеспечивается в методе App:Start() следующим образом:

oINI:= INIFileSpec("Printer Sample.INI")
Обратитесь к главе 12 для разъяснений по поводу класса INIFileSpec и скопируйте модуль C:INIFileSpec, разработанный в главе 12а, в нашу библиотеку. Вероятно, наилучшим способом этого будет экспортировать модуль из приложения 12а в файл MEF, а затем импортировать его в нашу библиотеку.

Теперь давайте рассмотрим метод ReportPrinter:Init() как наиболее важный метод этой главы. Первый параметр не нужен для целей печати вовсе, однако позднее мы добавим диалоговое окно AbortPrint (Прервать печать), для которого желательно наличие окна-владельца. Вторым параметром является JobName (Имя Задачи), назначение которого было объяснено выше. После выполнения некоторых проверок вызывается метод SUPER:Init() с параметрами JobName и <oPrintDev>, глобальной переменной, как уже упоминалось, присваивается значение в методе App:Start(), а посему объект становится доступным во всем приложении. Затем вызывается метод Printer:Idle(), который позволяет осуществлять печать в фоновом режиме. Далее мы получаем значения высоты и ширины страницы с помощью метода  Printer:CanvasArea() (доступ к соответствующей информации PrintingDevice осуществляется внутренними методами CA-Visual Objects).

Теперь следует определить 'Bounding Box' области печати внутри зоны, определяемой аппаратными особенностями принтера и нашими собственными границами полей, установленными в файле INI. Для этого мы вызываем Get_Margin(cEdge, lWidth), где <cEdge> является заданными значениями левого, правого, верхнего и нижнего полей, а логическое значение <lWidth> - TRUE или FALSE, в зависимости от полей, измеряемых в самой узкой части ширины листа, независимо от его ориентации. <oPrintDev:Device> возвращает строку, описывающую PrintingDevice, например, HP LaserJet 6L, таким образом:

oINI:GetInt(oPrintDev:Device, "Left")
возвратит значение "2", если файл INI соответствует приведенному выше. Это устанавливает <r4LeftMargin> в значение 2% от ширины страницы в пикселах. Для других полей - аналогично. Значения переменных <o_Origin> и <oDimension>, высчитываемые с учетом этих полей, определяют <oPrintArea> объекта Bounding Box, который сохраняется как свойство класса ReportPrinter. Этот объект должен рассчитываться чуть-чуть по-разному для разных вариантов ориентации листа бумаги. После этого <o_Origin> и <o_Dimension> могут быть уничтожены. Теперь мы можем быть уверены, что все координаты печати определены в терминах доступной зоны печати (PrintArea Bounding Box).

Мы сделаем только одно исключение из общего правила для печати внутри доступной зоны. Метод PrintFrame() напечатает четыре линии на самой границе аппаратно доступной зоны печати.

Метод Printer:Destroy() запускает печать, поэтому удостоверьтесь в его наличии, иначе результата не получите.

Следует более подробно объяснить работу метода PrintFrame(), поскольку это поможет нам в дальнейшем в объяснении использования LineObjects и других подобных методов. Он печатает четыре прямые линии. Массив <aFrame> содержит координаты четырех углов, с повторением первого угла для удобства. Значение "-1" обеспечит их печать внутри зоны печати.

Сначала создаются пустые объекты Point и Dimension. Объект Pen определяет толщину печатаемых линий. Объекты Point, Dimension и Pen загружаются в пустой, по существу, объект LineObject. Для каждой из четырех линий мы переопределяем содержание объекта LineObject и печатаем их на листе в цикле FOR…NEXT. В завершение всем объектам присваивается значение NULL_OBJECT, что позволяет функции CollectForced() удалить их из памяти.

Обратите внимание на то, что объекты Point имеют методы доступа по записи для значений X и Y. Аналогично, объекты Dimension имеют такие же методы для значений ширины и высоты, а метод LineObject() имеет такие же методы для Origin, Size и Pen. Это позволяет нам создавать только один объект каждого типа внутри метода и, тем самым, несколько экономить память. В данном конкретном случае это не имеет особого значения, однако если не применять подобные схемы, возможны ситуации, когда в памяти могут одновременно сосуществовать сотни объектов типа Line.

Прежде, чем продолжить, сохраните сделанное и создайте библиотеку, чтобы убедиться, что все идет по плану. Компиляция должна пройти успешно, без ошибок, но обратите внимание, что после появления надписи Build Done кнопка "Execute" на инструментальной панели остается отключенной. Это является следствием того, что мы создаем библиотеку, код которой может быть вызван только из какого-либо приложения.

Сетка - класс-наследник Printer

Мы создадим небольшой вспомогательный инструмент для раскладки страниц в классе ReportPrinter и его подклассах. Этот инструмент представляет из себя сетку, похожую на бумагу "в клеточку" со стороной ячейки в 1% от ширины и высоты зоны печати листа. Сетка имеет рамку, как описано выше, обрамляющую зону печати. Печатаемые линии имеют разную толщину, поскольку иначе из было бы трудно использовать. Вызвав метод PrintFrame(), рассмотренный выше, и PrintGrid(), можно напечатать набор горизонтальных и/или вертикальных линий.

Размещение линий определяется значениями X и Y в соответствующих одномерных массивах. Так, если массивы были {10,50,90} и {10,40,50}, результатом будут три горизонтальные, либо три вертикальные линии, либо их комбинация, как показано на рисунке 15.1, в зависимости от значения другого параметра. Местоположение вертикальных линий определяется значениями Х: 10%, 50% и 90% ширины зоны печати. Длина линий определяется самым маленьким и самым большим значением в массиве Y: 10 и 50. Размещение горизонтальных линий осуществляется аналогично. Таким образом, эти два массива создают сетку из вертикальных и горизонтальных линий.

PrintGrid() вызывается четыре раза в методе ReportExpose() класса OverlayGrid (см. ниже). Каждый вызов обращается к другому объекту Pen и другим массивам. В каждом случае используется тот же самый массив <aGrid> для величин X и Y, поскольку требования к сетке одинаковы в обоих измерениях.

Мы поместим код для OverlayGrid в модуле "C:OverlayGrid" в библиотеку, однако вызывать его будет, разумеется, из приложения.

Откройте редактор кода и введите:

CLASS OverlayGrid INHERIT ReportPrinter

METHOD Init(Owner) CLASS OverlayGrid

SUPER:Init(Owner, "Printing Overlay Grid")

RETURN SELF
METHOD ReportExpose() CLASS OverlayGrid

LOCAL aGrid AS ARRAY

LOCAL dwCounter AS DWORD

SELF:PrintFrame()

aGrid := {}

FOR dwCounter := 0 UPTO 100 STEP 50

AAdd(aGrid, dwCounter)

NEXT
SELF:PrintGrid(aGrid, aGrid, 5, "G", FALSE)

aGrid := {}

FOR dwCounter := 0 UPTO 100 STEP 10

AAdd(aGrid, dwCounter)

NEXT
SELF:PrintGrid(aGrid, aGrid, 3, "G", FALSE)

aGrid := {}

FOR dwCounter := 0 UPTO 100

AAdd(aGrid, dwCounter)

NEXT

SELF:PrintGrid(aGrid, aGrid, 1, "G", TRUE)

aGrid := {}

FOR dwCounter := 0 UPTO 100 STEP 5

AAdd(aGrid, dwCounter)

NEXT
SELF:PrintGrid(aGrid, aGrid, 1, "G", FALSE)

// SELF:Draw(RectangleObject{SELF:XYPoint(0, 0), SELF:WHDimension(1, 1), , Brush{BRUSHMEDIUM}})

RETURN FALSE
При первом вызове ReportPrinter:PrintGrid() массив <aGrid> имеет значения {0,50,100} в трех элементах и соответственно печатает три горизонтальных и три вертикальных линии в Х и У координатах 0, 50 и 100 процентов, как указано выше. Третьим параметром является размер Pen, а четвертым - буква 'G', указывающая на то, что требуется печать сетки (т.е. и горизонтальных, и вертикальных линий). Для горизонтальных линий используйте 'H', а для вертикальных - 'V'. Последний параметр определяет необходимость печати линий из множества точек, однако это возможно только при размере Pen равном 1.

При каждом последовательном вызове PrintGrid() значения в массиве и ширина линии изменяются. При третьем вызове последний параметр принимает значение TRUE, что означает необходимость печати строки с точками.

Вскоре мы увидим работу PrintGrid(), но предварительно необходимо обратить внимание на обращение к RectangleObject:Draw() непосредственно перед возвратом из метода. Обратите внимание на использование XYPoint() и WHDimension(). Здесь определяется затенение прямоугольника в начале зоны печати. Это просто выделяет ее начало на странице, чтобы было ясно - создана ли она при альбомной, либо при портретной ориентации листа. В данный момент эти строки закомментированы, поскольку нам еще нужно определить методы XYPoint() и WHDimension() для класса ReportPrinter.

Теперь перейдем к методу ReportPrinter:PrintGrid(), для чего следует создать новый модуль "C:ReportPrinter". Этот метод будет более общим, нежели просто класс OverlayGrid.

METHOD PrintGrid(aX AS ARRAY, aY AS ARRAY, dwPen AS DWORD,;

cType AS STRING, lDotted AS LOGIC) AS VOID CLASS ReportPrinter

LOCAL r4MinX, r4MaxX, r4MinY, r4MaxY AS REAL4

LOCAL oPen AS Pen

LOCAL oLine AS LineObject

LOCAL oPoint AS Point

LOCAL oDim AS Dimension

LOCAL dwCounter AS DWORD
IF ! SELF:NoPrint

cType := Pad(cType,1, " ")

IF ! (cType $ "GHV")

TextBox{ , "Не указано направление печати", "Должно быть 'G', 'H', или 'V'"}:Show()

ENDIF
r4MinX := r4MinY := 100

r4MaxX := r4MaxY := 0

AEval(aX, {| a | r4MinX := Min(r4MinX, a)})

AEval(aX, {| a | r4MaxX := Max(r4MaxX, a)})

AEval(aY, {| a | r4MinY := Min(r4MinY, a)})

AEval(aY, {| a | r4MaxY := Max(r4MaxY, a)})
IF lDotted

oPen := Pen{ , LINEDOTTED, dwPen}

ELSE

oPen := Pen{ , LINESOLID, dwPen}

ENDIF
oPoint := Point{0, 0}

oLine := LineObject{oPoint, oPoint, oPen}

oDim := Dimension{0, 0}

IF InStr(Upper(Left(cType, 1)), "HG")

FOR dwCounter := 1 UPTO ALen(aY)

oPoint:X := SELF:oPrintArea:Left + r4MinX * SELF:oPrintArea:Width / 100

oPoint:Y := SELF:oPrintArea:Bottom+aY[dwCounter] * SELF:oPrintArea:Height / 100

oDim:Height := 0

oDim:Width := (r4MaxX - r4MinX) * SELF:oPrintArea:Width / 100

oLine:Origin := oPoint

oLine:Size := oDim

oLine:Pen := oPen

SELF:Draw(oLine)

NEXT

ENDIF
IF InStr(Upper(Left(cType, 1)), "VG")

FOR dwCounter := 1 UPTO ALen(aX)

oPoint:X := SELF:oPrintArea:Left + aX[dwCounter] * SELF:oPrintArea:Width / 100

oPoint:Y := SELF:oPrintArea:Bottom + r4MinY * SELF:oPrintArea:Height / 100

oDim:Width := 0

oDim:Height := (r4MaxY - r4MinY) * SELF:oPrintArea:Height / 100

oLine:Origin := oPoint

oLine:Size := oDim

oLine:Pen := oPen

SELF:Draw(oLine)

NEXT

ENDIF
oPoint := NULL_OBJECT

oDim := NULL_OBJECT

oPen := NULL_OBJECT

oLine := NULL_OBJECT

CollectForced()

ENDIF
RETURN // NB : Метод VOID, поэтому никакого возвращаемого значения, даже NIL
Сначала об аргументах: <aX> содержит массив значений Х, определяющих горизонтальную позицию вертикальных линий, а <aY> содержит массив значений Y, определяющих вертикальные позиции горизонтальных линий, <dwPen> определяет толщину Pen, <cType> - тип сетки, а <lDotted> - необходимость печати точечной линии при ее ширине равной 1. <cType> может принимать значения 'G', 'H', 'V', что соответствует значениям "Сетка", "Горизонтальные линии", "Вертикальные линии".

Строка 
cType:= Pad(cType, 1, " ")
гарантирует, что в качестве <cType> передан, по крайней мере, один символ. В противном случае, если <cType> имеет значение NULL_STRING, метод выполнится некорректно.

Функция AEval() находит минимальное и максимальное значения в массивах <aX> и <aY> для определения длин строк, как было упомянуто выше.

Затем создается пустой объект LineObject, и, также как и в описанном PrintFrame(), объект наполняется несколько (или много) раз соответствующей информацией и выводится на печать методом Window:Draw(). В заключение некоторые объекты удаляются из памяти.

Теперь мы должны вызвать этот код из приложения, поэтому сохраните код, закройте редактор и скомпилируйте библиотеку, разработку которой мы завершили.

Тестовое приложение ReportPrinter

Для проверки библиотеки необходимо создать новое приложение, которое мы назовем "Ch15 App". Отметьте флажки Debug, GUI Classes, System Classes и RDD Classes и нажмите на кнопку ОК. Необходимо также добавить в состав приложения вновь разработанную нами библиотеку. Для этого откройте окно свойств приложения, выберите вкладку Libraries и в окне доступных (Available) библиотек вы увидите Ch15 Library. Скопируйте ее в окно включенных (Included) библиотек приложения и нажмите ОК.

Теперь мы напишем код для создания окна оболочки ShellWindow c меню, которое позволит нам распечатать OverlayGrid, созданную нами ранее. Весь код в библиотеке Ch15 Library доступен для нового приложения Ch15 App. 

Переименуйте модуль, созданный по умолчанию, в "C:Start" и введите следующий код:

METHOD Start() CLASS App

LOCAL oShell as ShellWindow
oPrintDev := PrintingDevice{}

oINI := INIFileSpec{""}

oShell := ShellWindow{SELF}

oShell:Caption := "Приложение ReportPrinter "

oShell:Menu := ShellMenu{oShell}

oShell:Show()

SELF:Exec()

oPrintDev := oINI := NULL_OBJECT

RETURN NIL
Обратите внимание на то, что <oPrintDev> и <oIni> не требуют объявления как локальные переменные, поскольку в библиотеке они объявлены как глобальные. Обе они уничтожаются после завершения работы приложения. Вскоре мы создадим и файл INI.

Теперь создайте другой модуль - "M:ShellMenu" и откройте редактор меню. Создайте меню с такими пунктами:

File

Exit

Reports

Overlay Grid

Windows
Help
Выберите пункт Exit и в окне его свойств в строке Event Name (имя события) введите "EndWindow". Аналогично в пункте меню Overlay Grid укажите имя события "OverlayGrid". Назовите меню "ShellMenu", сохраните его и закройте редактор.

Создайте еще один модуль, назовите его "C:ShellWindow", откройте редактор кода и введите следующий код:

METHOD OverlayGrid() CLASS ShellWindow

LOCAL oPrinter AS OverlayGrid
BEGIN SEQUENCE

oPrinter := OverlayGrid{SELF}

oPrinter:Start()

oPrinter:Destroy()

END SEQUENCE

RETURN NIL
Способ доступа к классу OverlayGrid и его коду, находящемуся в библиотеке, ничем не отличается от способа доступа к любому другому классу приложения. Конструкция BEGIN SEQUENCE … END SEQUENCE используется потому, что в методе ReportPrinter:Init() имеется предложение BREAK (См. Приложение 3). Такая конструкция будет использована во всех аналогичных методах ниже.

Теперь можно скомпилировать и запустить приложение. При выборе пункта меню Overlay Grid должна быть напечатана сетка 100х100. На листе не будет напечатана внешняя рамка, сетка должна занимать все аппаратно определенную область печати. Это происходит потому, что мы до сих пор не указали имя файла INI с соответствующим содержанием и не получили двухпроцентных полей, о которых говорилось выше. Обратите внимание на то, что внешняя строка сетки должна быть той же толщины, что и 50-процентные вертикальные и горизонтальные линии, однако реально она уже, потому что половина толщины линии выходит за пределы области печати.

Закройте приложение.

Откройте метод App:Start() и замените строку 

oINI:= INIFileSpec{""}
следующими строками:

oINI:= INIFileSpec{"Printer Sample.INI"}
IF ! File(oINI:FullPath)

oINI:WriteSection(oPrintDev:Device, {{"Left", "2"}, {"Right", "2"}, {"Top", "2"}, {"Bottom", "2"}})

oINI:WriteSection("Data Location", {{"DBF","C:\EVO\Video\"}})

ENDIF
SetDefault(oINI:GetString("Data Location", "DBF"))
Скомпилируйте и запустите приложение. Выберите печать сетки и теперь результат печати должен показать 2-процентные поля по всему периметру листа и рамку. Внешние линии сетки теперь должны быть верной толщины, поскольку область печати находится внутри аппаратно определенной области печати. <oPrintDev> принимает значение устройства печати ОС Windows по умолчанию, в авторском варианте, HP LaserJet 6L. Приведенный выше код создаст секцию в файле Printer Sample.ini с именем вашего устройства печати по умолчанию, полученного из <oPrintDev:Device>.

Мы установили путь по умолчанию к файлам данных на основании соответствующей строки в секции того же инициализационного файла. Если ваши файла находятся в другом каталоге, измените соответствующим образом строку пути.

Этот файл может иметь отдельные секции для каждого из устройств печати, имеющихся в сети, с соответственно установленными полями для каждого из них. Эти значения могут немного отличаться друг от друга, но конечный результат будет таким же. Секции для других принтеров могут быть добавлены в текстовом редакторе.

Итак, что делать с нашим расчерченным листком бумаги?  Прежде всего, можно начерно создать макет отчета в каком-либо текстовом процессоре. Затем можно напечатать его поверх уже имеющейся сетки и таким образом получить координаты для вывода элементов текста и графики в методе ReportExpose(). Постарайтесь запомнить, что размеры полей влияют на координаты размещения всех текстовых и графических объектов, за исключением рамки, и что поля могут быть изменены независимо от самого приложения. Ничего сверхъестественного, но иногда бывает полезно.

Интересно, что рамка, выводимая методом PrintFrame(), может слегка изменяться в зависимости от ориентации бумаги. Соответственно, несколько изменяется и область, занимаемая сеткой. Во избежание этой проблемы, всегда проверяйте правильность выбора типа сетки.

Остановка печати

До сих пор мы печатали только одну страницу. Если же нужно напечатать несколько страниц, было бы полезно видеть ход процесса и иметь возможность остановить печать. Мы изменим класс ReportPrinter таким образом, чтобы он показывал диалоговое окно с соответствующей информацией.

Рис. 15.2 взят из оконного редактора, на нем выделен элемент управления Fixed Text, используемый для отображения номера печатаемой страницы. Имя задачи добавляется в заголовок окна в методе PostInit().

Создайте новый модуль в библиотеке Ch15 Library и назовите его "W:AbortDialog", откройте диалоговое окно с именем AbortDialog в оконном редакторе и разметите в нем следующие элементы управления:

( Fixed Text в координатах 38, 50 и размером 93х20. Присвойте свойству Caption строку: "Печать страницы №:" В свойстве Generate Code укажите No.

( Fixed Text в координатах 130, 50 размером 60х20, присвоив ему имя (Name) "ftPage".

( PushButton в координатах 218, 50 и размером 80х20, присвоив ей имя (Name) "pbAbort", а свойству Caption - "Остановить".

Установите размеры окна: ширина - 336 и высота - 100. Присвойте свойству Caption окна строку "Printing", на вкладке DlgWindow установите свойство Modeless в <Yes>, а на вкладке Styles установите <Caption Bar Only>.

Сохраните окно, после чего вернитесь к вкладке DlgWindow. Щёлкните по свойству Class Declaration, чтобы открыть редактор кода и после надписи User Code Starts Here добавьте строку:

PROTECT oPrinter AS ReportPrinter
Сохраните сделанные изменения и закройте редактор. Перейдите к свойству PostInit Actions, откройте редактор кода и отредактируйте метод следующим образом:

METHOD PostInit(oParent, uExtra) CLASS AbortDialog

//Put your PostInit additions here

SELF:oPrinter := uExtra

SELF:Caption += ": " + SELF:oPrinter:JobName

RETURN NIL
Сохраните метод PostInit() и возвратитесь в оконный редактор. Выделите кнопку pbAbort и щелкните по свойству Click Event, вновь откроется редактор кода. Введите следующий код:

METHOD pbAbort() CLASS AbortDialog

SELF:oPrinter:Abort()

SELF:EndDialog(0)

RETURN NIL
Пропустите строку и напишите метод доступа по чтению и еще один метод:

ACCESS oDCftPage() CLASS AbortDialog

RETURN SELF:oDCftPage
METHOD Show() CLASS AbortDialog

SUPER:Show()

GetAppObject():Exec(EXECWHILEEVENT)

RETURN NIL
Сохраните код.

Это все, что необходимо для AbortDialog. Обратите внимание на то, что объект Printer передается диалоговому окну по ссылке, соответственно, при нажатии на кнопку "Остановить" можно послать сообщение объекту Printer о необходимости прекращения печати. При этом также закрывается диалоговое окно. Метод Show() необходим для того, чтобы организовать маленькую паузу за счет исполнения EXECWHILEEVENT. За это время ОС Windows успевает отобразить окно.

Закройте все редакторы и возвратитесь в репозиторий.

В модуле "C:ReportWriter" откройте объявление класса ReportPrinter. Необходимо добавить следующую строку в список переменных до объявления метода:

PROTECT oAbortDlg AS AbortDialog
Сохраните код, закройте редактор.

Откройте в редакторе кода метод Init() и вставьте две следующие строки непосредственно перед вызовом функции CollectForced() в конце метода:

SELF:oAbortDlg := AbortDialog{Owner, SELF}

SELF:oAbortDlg:Show()
Сохраните код, закройте редактор.

После объявления локальных переменных добавьте в метод ReportPrinter:PrinterExpose() следующую строку:

SELF:oAbortDlg:oDCftPage:Caption := NTrim(SELF:dwPage)
Это обеспечит обновление показаний счетчика для каждой страницы. Сохраните код, закройте редактор.

Откройте метод ReportPrinter:Start() и после обращения к IsMethod() на строке 9 добавьте ту же строку:

SELF:oAbortDlg:oDCftPage:Caption := NTrim(SELF:dwPage)
При желании можно сделать так, чтобы номера страниц, предшествующие требуемому диапазону, выводились красным цветом. Для этого вставьте приведенную ниже строку в метод ReportPrinter:Start() перед началам цикла DO WHILE:

SELF:oAbortDlg:oDCftPage:TextColor := Color{COLORRED}
Затем, после окончанием цикла ENDDO вставьте такую же строку, однако укажите цвет COLORBLACK для того, чтобы номера нужных страниц выводились черным цветом.

Сохраните код, закройте редактор.

Вновь откройте редактор кода и введите следующий код:

ACCESS JobName() CLASS ReportPrinter

RETURN SELF:sJobName
<sJobName> это свойство класса Printer. Сохраните код, закройте редактор.

Наконец, нам нужно добиться закрытия окна AbortDialog по окончании печати. Откройте метод ReportPrinter:Destroy() в редакторе и внесите необходимые исправления:

METHOD Destroy() CLASS ReportPrinter

Sleep(1000) 


// Новая строка

SELF:oAbortDlg:Destroy() 
// Новая строка 

SELF:oPrintArea := NULL_OBJECT

CollectForced()

SUPER:Destroy()

RETURN NIL
Сохраните код, закройте редактор и перекомпилируйте библиотеку.

Возвратитесь в приложение Ch15 App, перекомпилируйте и запустите его. Теперь во время печати сетки появляется окно AbortDialog, как и при печати любых других отчетов, если для этого используются классы, унаследованные от класса ReportPrinter.

Функция Sleep() используется потому, что некоторые задачи печати заканчиваются так быстро, что окно AbortDialog просто мелькает на экране,  в то время как сама печать занимает гораздо больше времени. (Если это не работает, убедитесь в том, что свойству окна Modeless присвоено значение <Yes>).

Как уже упоминалось, версии сетки для альбомной и портретной ориентации отличаются друг от друга. В настоящее время мы не можем проверить это утверждение, давайте возвратимся к этому позднее.

Настройка устройства печати

В классе PrintingDevice имеется метод Setup(), который отображает стандартный диалог настройки принтера. Обычно эта настройка доступна в меню "Файл". Откройте в редакторе меню ShellMenu и добавьте один пункт перед пунктом "Exit". Дайте ему имя "Printer Setup" ("Настройка печати"), а в качестве имени события Event Name укажите "PrinterSetup". Закройте редактор.

Теперь нам нужен метод PrinterSetup. Выберите модуль C:ShellWindow, откройте редактор кода и введите следующее:

METHOD PrinterSetup() CLASS ShellWindow

oPrintDev:Setup()

RETURN NIL
Сохраните код, закройте редактор, скомпилируйте и запустите приложение.

Выберите пункт меню <Printer Setup> из меню <File> и в диалоговом окне Print Setup Dialog установите альбомную ориентацию листа бумаги прежде, чем снова печатать сетку. Вероятнее всего, она будет напечатана несколько иначе, нежели при портретной ориентации, в зависимости от используемого принтера. Обратите внимание на то, что мы используем глобальную переменную GLOBAL <oPrintDev>, поэтому ориентация вывода относится ко всему приложению, ко всему, что может быть напечатано. Такое положение будет сохраняться до тех пор, пока вы не измените ее. Также имеется возможность программно установить ориентацию вывода, если, например, вы не желаете, чтобы пользователь изменял ее. Закройте приложение.

Пункт меню лучше именовать как Принтер, нежели Устройство печати, поскольку термин Принтер нам всем более знаком.

Методы масштабирования класса ReportPrinter
Возвратимся к библиотеке Ch15 Library. У нас есть еще многое, чем можно ее дополнить.

Нам необходимы еще два свойства в классе ReportPrinter. Выделите модуль C:ReportWriter, откройте объявление класса и дополните его следующими строками:

PROTECT r4VerticalPosition AS REAL4

PROTECT r4VerticalLineHeight AS REAL4

DECLARE METHOD Get_TextWidth

DECLARE METHOD WHDimension

DECLARE METHOD XYPoint
Первое из этих свойств определяет позицию для вывода различных объектов печати по вертикали. Это свойство может изменяться с помощью второго свойства во время печати строк текста.

Откройте редактор кода и создайте следующие методы доступа по чтению и записи для поддержки перечисленных выше переменных.

ACCESS dwPageHeight() CLASS ReportPrinter

RETURN SELF:dwPageHeight
ACCESS dwPageWidth() CLASS ReportPrinter

RETURN SELF:dwPageWidth
ACCESS r4VerticalLineHeight() CLASS ReportPrinter

RETURN SELF:r4VerticalLineHeight
ASSIGN r4VerticalLineHeight(n) CLASS ReportPrinter

SELF:r4VerticalLineHeight := n

RETURN n
ACCESS r4VerticalPosition() CLASS ReportPrinter

RETURN SELF:r4VerticalPosition
ASSIGN r4VerticalPosition(n) CLASS ReportPrinter

SELF:r4VerticalPosition := n

RETURN n
Прежде, чем закрыть редактор, пропустите строку и добавьте еще три метода, которые используют настройки Bounding Box в методе Init() как описано выше.

METHOD XYPoint(r4X AS REAL4, r4Y AS REAL4)AS Point CLASS ReportPrinter

RETURN Point{SELF:oPrintArea:Left + r4X * SELF:oPrintArea:Width / 100, ;

SELF:oPrintArea:Bottom + r4Y * SELF:oPrintArea:Height / 100}

METHOD WHDimension(r4W AS REAL4,r4H AS REAL4) AS Dimension CLASS ReportPrinter

RETURN Dimension{r4W * SELF:oPrintArea:Width / 100, r4H * SELF:oPrintArea:Height / 100}

METHOD Get_TextWidth(cString AS STRING) AS REAL4 CLASS ReportPrinter

LOCAL r4Width AS REAL4
r4width := 100 * SELF:SizeText(cString):Width / SELF:oPrintArea:Width

RETURN r4Width
Сохраните введенный код. Метод XYPoint() получает значения для X и Y в качестве параметров и возвращает объект Point со значениями, которые "понимает" устройство печати. То есть XYPoint(10,30) возвратит объект Point со значениями X и Y равными 380 и 1223, однако, к счастью, нам лично не придется иметь дело с этими значениями, мы будем иметь дело лишь со значениями 10 и 30, независимо от размеров бумаги и разрешающей способности принтера.

Аналогично метод WHDimension() определяет объект Dimension, используя нашу систему координат, в простых терминах ширины и высоты.

Метод Get_TextWidth() позволяет определить ширину любого текста как отношение длины строки к ширине области печати (принимается за 100%). В дальнейшем, при разработке выравнивания текста и переноса текста по словам, этот метод окажет нам неоценимую помощь.

Закройте редактор и откройте метод OverlayGrid:ReportExpose(). Теперь мы может убрать оператор комментария из строки ближе к концу метода, которая позволяет печатать небольшие прямоугольные объекты. Он показывает начало сетки, в левом нижнем углу, и, таким образом, показывает, была ли сетка напечатана в портретной или альбомной ориентации.

Сохраните код, закройте редактор, и скомпилируйте библиотеку. Вернитесь к приложению Ch15 App, перекомпилируйте его и запустите для проверки правильности выполненной работы.

Выравнивание текста

До сих пор мы не напечатали ни единой строки текста. Далее мы обнаружим, что существуют несколько категорий текста, которые нужно печатать как индивидуальные текстовые объекты, колонки текста, заметки (мемо) или свободный текст.

Для печати текста можно использовать два метода, предоставляемые ОС Windows: Draw(oTextObject) и TextPrint(cText, oPoint). Мы используем последний из них. Однако, этот метод не обеспечивает выравнивание текста ни по левой стороне, ни по правой, ни по центру, поэтому нам придется обеспечить это самим. Вместо переработки метода TextPrint представляется более удобным создать новый метод, который мы назовем PrintText(). Он используется следующим образом:

SELF:PrintText(0, "Company name", ALIGN_LEFT)

// измените шрифт, если требуется

SELF:PrintText(50, "Title wording", ALIGN_CENTRE)

// верните прежний шрифт
SELF:PrintText(100, "Printed on: " + DTOC(Today()), ALIGN_RIGHT)
В результате будет напечатано нечто вроде - 

Company name 

Title wording 

Printed on: 13/04/99
в текущей позиции <ReportPrinter:r4VerticalPosition>.

Перейдите в модуль C:ReportWriter, откройте редактор кода и введите код метода PrintText():

METHOD PrintText(X AS REAL4,cText AS STRING,;

dwAlign AS DWORD) AS VOID CLASS ReportPrinter

LOCAL r4Len AS REAL4

IF ! SELF:NoPrint

IF dwAlign = ALIGN_LEFT

SELF:TextPrint(cText, SELF:XYPoint(X, SELF:r4VerticalPosition))

ELSE

r4len := SELF:Get_TextWidth(cText)

IF dwAlign = ALIGN_CENTRE

SELF:TextPrint(cText, SELF:XYPoint(X - r4Len / 2, SELF:r4VerticalPosition))

ELSE

SELF:TextPrint(cText, SELF:XYPoint(X - r4Len, SELF:r4VerticalPosition))

ENDIF

ENDIF

ENDIF

RETURN // NB : Метод VOID, поэтому нет возвращаемого значения, даже NIL.

DEFINE ALIGN_LEFT := 1
DEFINE ALIGN_CENTRE := 2

DEFINE ALIGN_RIGHT := 3
Как видите, метод PrintText() использует методы XYPoint() и Get_TextWidth() класса ReportPrinter. Манифестные константы используются для того, чтобы сделать код более читаемым. Сохраните код, закройте редактор.

Метод PrintText() теперь должен быть объявлен также, как и другие типизированные методы в объявлении класса, следовательно, откройте определение класса и добавьте в него строку:

DECLARE METHOD PrintText
Сохраните код, закройте редактор. Этот метод будет использован в следующей части нашей главы для вывода заголовка к заметке.

Перенос слов в тексте

Давайте разберемся, как нам напечатать текст. Этот текст может быть неформатированным текстовым файлом, либо отдельной строкой. Проблемы очень схожи, но в данной главе мы разберем только печать строки текста. Во всем тексте будет использован единый шрифт, однако в тексте могут присутствовать маркеры параграфов и конца страницы. Мы должны будем определить, сколько слов может разместиться в одной строке печатного текста, принимая в расчет разрывы строк и страниц, кроме того, мы должны уметь делать все перечисленное, независимо от используемого шрифта. В то время как сам текст строки не позволяет использовать разные шрифты одновременно, у нас будет еще заголовок, в котором может использоваться шрифт, отличный от общего.

В библиотеке Ch15 Library создайте новый модуль и озаглавьте его C:ReportMemo. Откройте редактор кода и импортируйте код из файла Code2.prg с диска CD.

CLASS ReportMemo INHERIT ReportPrinter

PROTECT oHeader AS ReportMemoHeader

PROTECT cMemo AS STRING

PROTECT dwPointer AS DWORD

PROTECT dwMaxChrs AS DWORD
METHOD Init(Owner, cJobName, cString) CLASS ReportMemo

Default(@cString, NULL_STRING)

IF SLen(cString) = 0

ErrorBox{ , "Пустая строка!"}:Show()

BREAK

ELSE

SUPER:Init(Owner, cJobName)

SELF:Owner := Owner

SELF:cMemo := cString 

SELF:Font := Font{FONTSWISS10}

SELF:dwMaxChrs := DWORD(500 / SELF:Get_TextWidth("iiiii"))

SELF:r4VerticalLineHeight := 1.5

SELF:dwPointer := 1

SELF:oHeader := ReportMemoHeader{SELF}

ENDIF

RETURN SELF
METHOD Destroy() CLASS ReportMemo

SELF:oHeader := NULL_OBJECT

SUPER:Destroy()

RETURN NIL
METHOD ReportExpose() CLASS ReportMemo

LOCAL cLine, cCR, cLF, cLFCR, cEJECT AS STRING

LOCAL dwLineLen, dwPos, dwPos1, dwPos2 AS DWORD

LOCAL lEject AS LOGIC

SELF:oHeader:Print()
DO WHILE SELF:r4VerticalPosition >= 2

cLine := SubStr3(SELF:cMemo, SELF:dwPointer, SELF:dwMaxChrs)

dwLineLen := 0

IF (dwPos1 := At2(CRLF, cLine)) > 0 .OR. (dwPos2 := At2(cLFCR, cLine)) > 0

dwPos := Min_Non0(dwPos1, dwPos2)

dwLineLen := 2

ELSEIF (dwPos1 := At2(cLF, cLine)) > 0 .OR. (dwPos2 := At2(cCR, cLine)) > 0

dwPos := Min_Non0(dwPos1, dwPos2)

dwLineLen := 1

ENDIF
IF dwPos > 0

cLine := SubStr3(cLine, 1, dwPos - 1)

ENDIF
IF (dwPos := At2(cEJECT, cLine)) > 0

dwLineLen := 1

lEject := TRUE

cLine := SubStr3(cLine, 1, dwPos - 1)

ENDIF
DO WHILE SELF:Get_TextWidth(cLine) > 100

dwPos := RAt2(" ", cLine)

IF dwPos > 0

cLine := SubStr3(cLine, 1, dwPos -1)

dwLineLen := 1

ELSE

dwPos := RAt2("-", cLine)

IF dwPos > 0

cLine := SubStr3(cLine, 1, dwPos -1)

dwLineLen := 1

ELSE

cLine := SubStr3(cLine, 1, SLen(cLine) - 1)

dwLineLen := 0

ENDIF

ENDIF

ENDDO

dwLineLen += SLen(cLine)

SELF:dwPointer += dwLineLen

SELF:PrintText(0, cLine, ALIGN_LEFT)

SELF:r4VerticalPosition -= SELF:r4VerticalLineHeight

IF lEject

lEject := FALSE

EXIT

ENDIF

ENDDO

RETURN SELF:dwPointer < SLen(SELF:cMemo)
DEFINE cCR := _Chr(13)

DEFINE cLF := _Chr(10)

DEFINE cLFCR := _Chr(10) + _Chr(13)

DEFINE cEJECT := _Chr(12)

FUNCTION Min_Non0(dwA AS DWORD, dwB AS DWORD) AS DWORD

LOCAL dwRet AS DWORD

IF Empty(dwA)

dwRet := dwB

ELSEIF Empty(dwB)

dwRet := dwA

ELSE

dwRet := Min(dwA, dwB)

ENDIF

RETURN dwRet
Сохраните код.

В методе инициализации прежде всего мы должны проверить, что нам не передана пустая строка. Предполагая, что с этим все в порядке, далее мы должны определить наибольшее число самых узких символов, которые могут уместиться на странице (прим. пер. - так в тексте. Вероятно, авторы имели в виду количество символов в одной строке, о чем они и говорят далее). Предположим, что наиболее узким символом является буква "i" и с учетом этого подсчитаем <SELF:dwMaxChrs>. Это даст нам максимальное число символов, которые могут быть напечатаны в каждой из строк текста.

Созданным нами объект Header должен быть уничтожен, что делается в методе Destroy(). Обратите внимание на то, что объект Header уничтожается до разрушения ReportMemo, в противном случае разрушится SELF, и, соответственно, переменная <oHeader> окажется недоступной. 

Если посмотреть в метод ReportExpose(), то мы увидим, что <SELF:dwMaxChrs> используется для извлечения числа символов с самого начала текста, что позволяет нам иметь дело только с одной строкой, независимо от их общего количества.

Далее мы должны обнаружить маркер конца строки. Эти маркеры представляют собой различные комбинации символов Chr(10) и Chr(13), в зависимости от редактора, в котором была создана строка. При обнаружении такого маркера мы должны сначала определить, является он односимвольным (Chr(10)) или же двухсимвольным (Chr(13)+Chr(10)), а затем "отрезать" часть строки до маркера, не обращая внимания на все то, что осталось за маркером. Однако, в оставшейся части может содержаться маркер перевода страницы (Chr(12)). Соответственно, мы должны "запомнить", что после печати строки требуется выполнить переход на новую страницу.

Теперь, когда длина строки уменьшена до максимального числа символов <SELF:dwMaxChrs>, она все равно может оказаться слишком длинной для того, чтобы уместиться в одной строке на листе бумаги, поскольку в строке содержатся и более широкие символы. Поэтому нужно последовательно уменьшать длину строки, начиная справа, путем ее обрезания при нахождении символов переноса или пробелов до тех пор, пока длина строки не уменьшится до требуемой. (При этом флаг перехода на новую страницу должен быть сброшен). Если же в строке отсутствуют и переносы, и пробелы, она все равно должна быть обрезана так, чтобы уместиться на странице.

В конце концов мы получили первую строку для вывода на печать. Мы также определили общее число печатаемых и интерпретируемых символов, например, Chr(13), которые не печатаются. Эта сумма печатаемых и непечатаемых символов, <dwLineLen>, в дальнейшем используется для модификации указателя <SELF:dwPointer> для определения точного места в тексте, с которого должно начинаться извлечение следующей строки. Далее мы используем метод PrintText() для печати строки.

В завершение мы уменьшаем значение <SELF:r4VerticalPosition> на величину <SELF:r4VerticalLineHeight> для подготовки к печати следующей строки.

При необходимости перехода на новую страницу мы выходим из цикла, в противном случае его выполнение продолжается. Неоднократно упоминавшийся переход на новую страницу управляется специальным символом, находящимся в тексте. Однако, независимо от его наличия и местоположения, мы должны обеспечить переход на новую страницу после печати последней строки листа. Для этого мы должны определить, остался ли еще текст для его печати на следующей странице. Это достигается сравнением общего размера текста с <SELF:dwPointer>.
Метод ReportExpose() обрабатывает текст с учетом высказанных выше замечаний и печатает строки в цикле DO WHILE … ENDDO до тех пор, пока этот цикл не определит, что вывод на печать осуществляется вблизи нижней границы области печати листа. Внутренний цикл DO WHILE … ENDDO обрезает строку до тех пор, пока она не сократится настолько, чтобы уместиться на листе, с учетом того, что она может быть уже достаточно коротка, если ранее был обнаружен один из маркеров - конца строки или перехода на новую страницу.

Приведенный выше код предполагает, что CRLF и <cLFCR> не могут одновременно встречаться в тексте. Обратите внимание на то, что _Chr() может быть обработана препроцессором во время компиляции, и, следовательно, работает быстрее, нежели аналогичный вызов Chr(). В CA-Visual Objects константа CRLF определена как _Chr(13)+_Chr(10), поэтому нам не нужно дополнительно определять ее.

Функция Min_Non0() возвращает наименьшее значение из двух ненулевых параметров или сам ненулевой параметр, если существует только один из них.

Теперь нам нужно определить заголовок для каждой страницы. Метод Print() для заголовка вызывается лишь один раз на страницу из метода ReportExpose(). В силу природы печати страницы не имеет значения, где именно на ней располагаются объекты печати, поэтому можно обеспечить печать и подножия каждой страницы (при наличии необходимой информации). Например, без каких-либо затруднений можно обеспечить печать текущего номера страницы вверху и внизу листа. Этот объект создается внутри метода Init() класса ReportMemo. Скопируйте приведенный ниже код в редактор кода, пропустив одну строку после функции Min_Non0():

CLASS ReportMemoHeader

PROTECT Owner AS ReportMemo
METHOD Init(owner) CLASS ReportMemoHeader

SELF:Owner := Owner

RETURN SELF
METHOD Print() CLASS ReportMemoHeader

SELF:Owner:r4VerticalPosition := 99.0

SELF:Owner:PrintText(0, "Memo Print", ALIGN_LEFT)

SELF:Owner:PrintText(100, DToC(Today()), ALIGN_RIGHT)

SELF:Owner:PrintGrid({0,100}, {98.5, 98.5}, 2, "H", FALSE)

SELF:Owner:r4VerticalPosition := 0.0

SELF:Owner:PrintText(50, NTrim(SELF:Owner:dwPage), ALIGN_CENTRE)

SELF:Owner:r4VerticalPosition := 96.0

RETURN NIL
Сохраните код, закройте редактор и скомпилируйте библиотеку.

Чтобы увидеть результаты своего труда, нужно немного изменить код приложения Ch15 App. Откройте ShellMenu в редакторе меню и после элемента Overlay Grid добавьте еще один элемент в меню Reports. Его свойству Caption присвойте значение "Print a Memo (String)", а свойству Event Name - "PrintString".
Сохраните сделанные изменения и закройте редактор.

Перейдите к модулю C:ShellWindow, откройте редактор кода и введите следующий код:

METHOD PrintString() CLASS ShellWindow

LOCAL cFile, cString AS STRING

LOCAL oPrinter AS ReportMemo
cFile := "Printer Sample.INI"

IF File(cFile)

cString := MemoRead(cFile)

BEGIN SEQUENCE

oPrinter := ReportMemo{SELF, cFile, cString, "", cFile}

oPrinter:Start()

oPrinter:Destroy()

END SEQUENCE

ENDIF

RETURN NIL
Как вы видите, мы получили строку, загрузив ее с помощью функции MemoRead(). Затем мы можем напечатать его.

Сохраните код, закройте редактор, скомпилируйте и запустите приложение. Выберите пункт меню <Print a Memo (String)>, который выведет на печать содержание файла Printer Sample.ini. Закройте приложение.

Разработанный нами код работает неплохо при длине строки, скажем, до 100 Кб. Однако, на практике пользователи ваших программ могут пожелать напечатать файл размером в несколько мегабайт. Код, находящийся на CD, обеспечивает печать как строк, так и файлов, но он слишком велик для детального рассмотрения здесь. Кроме того, различия не настолько уж существенны ни с точки зрения объектно-ориентированного программирования, ни с точки зрения реализации класса Printer. Позднее вы сможете импортировать библиотеку ReportPrinter с диска, а пока оставайтесь с нами!

Колонки

Большинство приложений, работающие с базами данных, требуют создания отчетов в табличной форме. И вновь мы можем помочь себе сами, создав специальные объекты для вывода данных в виде колонок. Мы создадим класс колонки - ReportCol и создадим столько объектов колонок, сколько нам нужно, сохраняя ссылки на них в объекте PrinterObject для отчета, как мы делали для заголовка Header в ReportMemo выше.

Для каждой колонки нам требуется знать ее смещение относительно левого поля, ее ширину и характер выравнивание текста внутри нее. В дальнейшем мы модифицируем код таким образом, чтобы колонки могли содержать общие и промежуточные суммы по соответствующим строкам колонки.

Класс ReportCol и его методы будут созданы в том же модуле, что и класс ReportPrinter. В библиотеке Ch15 Library откройте редактор кода и введите следующее:

CLASS ReportCol

PROTECT Owner AS OBJECT

PROTECT cOldTextValue AS STRING

PROTECT r4XPos AS REAL4

PROTECT r4Width AS REAL4

PROTECT dwALIGN AS DWORD
METHOD Init(Owner, cTextValue, lNumeric, r4XPos, r4Width, ;

lRepeat, dwAlign, aTotals) CLASS ReportCol

Default(@r4XPos, 10)

Default(@r4Width, 500)

Default(@dwAlign, 1)

SELF:Owner := Owner

SELF:cOldTextvalue := ""

SELF:r4XPos := r4XPos

SELF:r4Width := r4Width

SELF:dwALIGN := dwAlign

RETURN SELF
ACCESS cOldTextValue() CLASS ReportCol

RETURN SELF:cOldTextValue
METHOD Print(uPrint) CLASS ReportCol

LOCAL cText AS STRING

LOCAL r4X, r4Y, r4Z AS REAL4

r4X := SELF:r4XPos

r4Y := SELF:Owner:r4VerticalPosition

r4Z := SELF:Owner:r4VerticalLineHeight

IF UsualType(uPrint) = VOID

cText := SELF:cOldTextValue

ELSEIF ValType(uPrint) == "C"

cText := uPrint

ENDIF
IF SELF:dwALIGN = ALIGN_RIGHT

r4X += SELF:r4Width

ENDIF
IF SELF:dwALIGN = ALIGN_CENTRE

r4X += SELF:r4Width / 2

ENDIF

SELF:Owner:PrintText(r4X, cText, SELF:dwAlign)

RETURN NIL
Сохраните код и закройте редактор.

Некоторые временные "ложные" аргументы (<lNumeric>, <lRepeat>, <aTotals>) включены в метод Init() для того, чтобы не вставлять их позднее, но сейчас они ничего не делают. Переменная <cOldTextValue> названа именно так, поскольку позднее мы добавим переменную <cNewTextValue>. Владельцем объекта Column всегда будет объект ReportPrinter, поскольку будет требоваться доступ к его свойствам. Мы будем использовать объекты ReportCol в объектах Header, Body и Total и владельцем всех их будет ReportPrinter.

Метод ReportCol:Print() включает в себя код для выравнивания текста в колонке влево, вправо или по центру, в отличие от метода ReportPrinter:PrintText(), который обеспечивал выравнивание текста применительно к горизонтальной позиции, тем не менее, этот метод используется внутренне. Если текст слишком широк для колонки, он не обрезается.

Скомпилируйте библиотеку, чтобы удостовериться, что все сделано правильно.

Многоколоночные списки

Для начала мы используем объекты колонки в ситуации, когда суммы не требуются, поскольку такой код у нас еще отсутствует. Хорошим примером будет список типа "Телефонный каталог", в котором данные организованы в несколько колонок, каждая из которых представляет из себя список строк данных, следующих друг за другом. Результат выглядит примерно так, как указано на рис. 15.3, с тем лишь отличием, что обычно первая колонка доходит до конца страницы прежде, чем начнется другая колонка.

Код, который мы создадим, будет весьма прост в использовании: нужно лишь передать в качестве параметров Server (с предварительно подключенным индексом и фильтром), некоторую информацию о полях и число колонок (в данном случае 3). Информация о полях суть массив имен полей, пропорциональные значения ширины колонок, а также сведения о выравнивании. Все данные должны поступать от одного сервера.

Тот факт, что формат списка повторяется на всей странице, не очень сильно усложняет код, однако создает определенные трудности, посему разобьем задачу на небольшие этапы. Создайте новый модуль C:ReportColList в библиотеке Ch15 Library, откройте редактор кода и импортируйте в него код из файла Code3.prg, находящегося на CD.

CLASS ReportColListPrint INHERIT ReportPrinter

PROTECT oServer AS DBServer

PROTECT aBndgBox AS ARRAY

EXPORT aFields AS ARRAY

EXPORT aCols AS ARRAY

PROTECT oHeader AS ReportColListHeader
METHOD Init(Owner, cJobName, oServer, aFields, dwBoxes) CLASS ReportColListPrint

LOCAL dwCounter AS DWORD

LOCAL r4TotWidth, r4Offset AS REAL4

LOCAL aColPosns, aColWidths AS ARRAY

LOCAL oCol AS ReportCol

SUPER:Init(Owner, cJobName)

Default(@dwBoxes, 1)

SELF:Owner := Owner

SELF:oServer := oServer

SELF:aFields := aFields

SELF:aCols := {}

SELF:oHeader := ReportColListHeader{SELF}

r4TotWidth := 0

FOR dwCounter := 1 UPTO ALen(aFields)

r4TotWidth += aFields[dwCounter, 2]

NEXT
aColPosns := {0}

r4Offset := 0
FOR dwCounter := 1 UPTO ALen(aFields)

r4Offset += (aFields[dwCounter,2] * 100) / r4TotWidth

AAdd(aColPosns, r4Offset)

NEXT
aColWidths := {}
FOR dwCounter := 1 UPTO ALen(aFields)

AAdd(aColWidths, aColPosns[dwCounter + 1] - aColPosns[dwCounter])

NEXT
FOR dwCounter := 1 UPTO ALen(aFields)

oCol := ReportCol{SELF, "", FALSE, aColPosns[dwCounter], aColWidths[dwCounter], TRUE, ;

aFields[dwCounter,3]}

AAdd(aCols, oCol)

NEXT
oCol := NULL_OBJECT

SELF:Font := Font{FONTSWISS10}

SELF:r4VerticalLineHeight := 1.5

RETURN SELF

METHOD Destroy() CLASS ReportColListPrint

LOCAL dwCounter AS DWORD

SELF:oHeader := NULL_OBJECT

FOR dwCounter := 1 UPTO ALen(SELF:aBndgBox)

SELF:aBndgBox[dwCounter] := NULL_OBJECT

NEXT

SUPER:Destroy()

RETURN NIL

METHOD ReportExpose() CLASS ReportColListPrint

LOCAL dwCounter AS DWORD

LOCAL cText AS STRING

SELF:oHeader:Print()

SELF:r4VerticalPosition := 95.0

DO WHILE SELF:r4VerticalPosition > 3.0 .AND. ! SELF:oServer:EOF

FOR dwCounter := 1 UPTO ALen(SELF:aFields)

cText := SELF:oServer:FIELDGET( SELF:aFields[dwCounter, 1])

SELF:aCols[dwCounter]:Print(cText)

NEXT
SELF:r4VerticalPosition -= SELF:r4VerticalLineHeight

SELF:oServer:Skip()

ENDDO
RETURN ! SELF:oServer:EOF
CLASS ReportColListHeader

PROTECT Owner AS ReportColListPrint

METHOD Init(Owner) CLASS ReportColListHeader

SELF:Owner := Owner

RETURN SELF

METHOD Print() CLASS ReportColListHeader

LOCAL dwCounter AS DWORD

LOCAL cText AS STRING

SELF:owner:r4VerticalPosition := 98.0

FOR dwCounter := 1 UPTO ALen(SELF:Owner:aFields)

cText := SELF:Owner:aFields[dwCounter,4]

SELF:Owner:aCols[dwCounter]:Print(cText)

NEXT
SELF:Owner:PrintGrid({0, 100}, {97,97}, 2, "H", FALSE)

RETURN NIL
Сохраните код.

Значительная часть кода метода Init() посвящена определению позиций колонок и ширины каждой из них применительно к полям, перечисленным в <aFields>. Каждый элемент этого массива представляет из себя четырехэлементный массив, содержащий:

· имя поля данных для колонки;

· долю ширины страницы, выделенную для данного поля;

· информацию о выравнивании (влево, вправо, по центру);

· заголовок колонки.

Хотя высота строки VerticalLineHeight жестко закодирована в методе Init(), она может быть переопределена в наследуемых классах.

Как обычно, мы будем должны уничтожить все созданные объекты, это делается в методе ReportColListPrint:Destroy().

Метод ReportExpose() (см. ниже) циклически проходит по всему серверу, начиная с текущей позиции указателя до тех пор, пока VerticalPosition не станет слишком маленькой, когда будет достигнут конец страницы. Для каждой записи он вызывает метод ReportCol:Print() для каждого поля, указанного в массиве <aFields>.

Метод Print() класса ReportColListHeader обеспечивает печать имен полей и их подчеркивание в верхней части страницы.

Закройте и перекомпилируйте библиотеку. Внесите необходимые изменения в приложение Ch15 App. Откройте ShellMenu в редакторе. Добавьте в меню Reports элемент с именем "List Columns of Male Stars", а свойству Event Name присвойте значение "ListColumns". Сохраните меню и закройте редактор.

Перейдите в модуль, откройте редактор кода и введите следующий код:

METHOD ListColumns() CLASS ShellWindow

LOCAL oPrinter AS ReportColListPrint

LOCAL aFields AS ARRAY

LOCAL oServer AS DBserver

oServer := StarM{}

aFields := { {"FName", 10, ALIGN_LEFT, "First Name"}, {"LName", 10, ALIGN_LEFT, "Last Name"}, ;

{"RefNo", 5, ALIGN_RIGHT, "Ref No"}}

BEGIN SEQUENCE

oPrinter:=ReportColListPrint{SELF, "List Printing", oServer, aFields}

oPrinter:Start()

oPrinter:Destroy()

END SEQUENCE

RETURN NIL
Как  видно из приведенного кода, в нем открывается сервер StarM и создается массив с именами нужных полей, пропорциональной шириной каждой колонки и значением выравнивания. Ширина каждого из полей пропорциональна, поэтому поля имени и фамилии будут вдвое шире поля RefNo (ширина каждого из них равна 10, 10 и 5, соответственно). Четвертым элементом каждого из подмассивов является заголовок для данного поля.

Сохраните код, закройте редактор.

Теперь нам нужен сервер. Выделите проект EVO в левой панели репозитория, импортируйте файл Ch15 Servers.AEF с диска CD и скомпилируйте его.

Для того, чтобы включить эту библиотеку в приложение, возвратитесь к Ch15 App, откройте окно свойств приложения, щелкните по вкладке Libraries и скопируйте библиотеку из окна Available в окно Included. Закройте окно, скомпилируйте и запустите приложение. Выберите пункт меню List Columns of Male Star и вы должны получить трехколоночный отчет с полями: Имя, Фамилия и Ссылочный № (First Name, Last Name и Ref №). Закройте приложение.

Следующей задачей будет сделать список поуже таким образом, чтобы можно было повторять его на странице, как это делается в телефонных справочниках. Мы можем сделать это относительно легко, временно изменив границы области печати, определенную ReportPrinter:PrintArea, во время печати каждой из колонок. Действительная ширина колонок определена нами пропорционально, это будет работать и при более узкой области печати. Нам необходимо создать массив значений BoundingBox внутри класса ReportColListPrint, определить их в методе Init(), а затем использовать их в методе ReportExpose().

В модуле Ch15 Library откройте метод ReportColListPrint:Init() в редакторе кода и вставьте строку

LOCAL r4Width AS REAL4,

а затем, непосредственно перед возвратом из метода, вставьте:

SELF:aBndgBox := {}

r4Width :=SELF:oPrintArea:Width / (1.08 * dwBoxes - 0.08)

FOR dwCounter := 1 UPTO dwBoxes

AAdd(SELF:aBndgBox, BoundingBox{Point{0, 0}, Point{0 ,0}})

SELF:aBndgBox[dwCounter]:Bottom:=SELF:oPrintArea:Bottom

SELF:aBndgBox[dwCounter]:Left := SELF:oPrintArea:Left + 1.08 ;

* (dwCounter - 1) * r4Width

SELF:aBndgBox[dwCounter]:Height:=SELF:oPrintArea:Height

SELF:aBndgBox[dwCounter]:Width :=r4width

NEXT
Этот код создает отдельные BoundingBoxes с одинаковыми значениями вертикального размещения и высотой, однако с разными значениями Left и Right. Если потребуется оригинальный вариант BoundingBox, тогда массив должен содержать только один элемент. Промежуток между BoundingBoxes жестко закодированы и определен в 8% от ширины каждой.

Сохраните код и закройте редактор.

Откройте метод ReportExpose() и измените его следующим образом:

METHOD ReportExpose() CLASS ReportColListPrint

LOCAL dwCounter, dwCounter2 AS WORD 


// Изменено
LOCAL cText AS STRING
FOR dwCounter2 := 1 UPTO ALen(SELF:aBndgBox) 
// Новая строка
SELF:oPrintArea := SELF:aBndgBox[dwCounter2]
// Новая строка
SELF:oHeader:Print()

SELF:r4VerticalPosition := 95.0

DO WHILE SELF:r4VerticalPosition > 3.0 .AND. ! SELF:oServer:EOF

FOR dwCounter := 1 UPTO ALen(SELF:aFields)

cText := SELF:oServer: FIELDGET(SELF:aFields[dwCounter, 1])

SELF:aCols[dwCounter]:Print(cText)

NEXT

SELF:r4VerticalPosition -= SELF:r4VerticalLineHeight

SELF:oServer:Skip()

ENDDO
NEXT 






// Новая строка

RETURN ! SELF:oServer:EOF
Теперь метод ReportExpose() проходит по каждой из BoundingBox, заполняя их необходимыми данными.

Сохраните код, закройте редактор.

ReportColListHeader не требует никаких изменений. Теперь вы увидите, почему Header не печатает номера страниц - он будет использоваться несколько раз на одной странице.

Скомпилируйте библиотеку и возвратитесь в Ch15 App.

В модуле C:ShellWindow откройте в редакторе метод ListColumns(). Добавьте 5-й параметр со значением 3 в вызов ReportColListPrint() для использования нового многоколоночного списка. 

oPrinter:= ReportColListPrint{SELF, "List Printing", oServer, aFields, 3}
Сохраните код, закройте редактор, скомпилируйте и запустите приложение. Повторите последний тест и убедитесь в том, что теперь заголовки повторяются на странице в нужном виде, а выравнивание текста в колонках не претерпело изменений.

Все записи из файла выводятся в крайней левой колонке. Для того, чтобы распечатать их в трех колонках, как показано на рис. 15.3, временно измените нижнее поле с 3.0 на 88.0 в строке 8 метода ReportExpose() метода ReportColListPrint. Перекомпилируйте библиотеку и приложение, запустите его и, выбрав пункт меню List Columns of Male Star, напечатайте многоколоночный отчет.

Мы еще далеко не использовали все возможности класса ReportCol. Мы возвратимся к нему позднее.

Эмуляция QOut() в классе Printer
Функция QОut() (более известен ее командный эквивалент ?), была весьма удобна при работе в Клиппере. Она никогда не создавала впечатляющих визуальных эффектов, но она может оказаться полезной и при работе в CA-Visual Objects. Она до сих пор доступна, если в приложение включена библиотека Terminal Lite. Наибольшим разочарованием при работе с терминальным окном является то, что открыв его, закрыть его уже невозможно. Его можно только свернуть. Мы можем придумать вариант получше, хотя нам и придется иметь дело с единственным шрифтом.

Новая версия QOut() будет сохранять текст, а затем все сохраненное будет выводиться в файл, либо на печать, либо и в файл, и на печать.

Можно сразу увидеть несколько возможных применений для этой функции. В ходе работы приложения можно сохранять некоторые части текста, которые удовлетворяют некоторым условиям.

Представляемый далее код можно лишь частично считать эмуляцией функций QOut() и QQOut(), поскольку функции ROut() и RROut() принимают список с неопределенным количеством параметров разных типов. (Буква R в названии функций выбрана лишь потому, что она следует за Q). Допустимыми типами данных являются все типы, которые могут быть конвертированы функцией AsString(), обычно это числовой тип данных, строка, дата и логический. Все они конвертируются в строки перед выводом. Вывод использует код, рассмотренный ранее, в модуле C:ReportMemo библиотеки Ch15 Library. Таким образом, наша цель сейчас - собрать данные для вывода их с помощью метода ReportMemo. 

Связь с другими частями приложения осуществляется с помощью четырех функций: ROutStart(), ROut(), RROut() и ROutEnd(). Первая подготавливает контейнер для текста, будь то файл или строка, последняя выводит его на экран, либо сохраняет его в файле, две оставшиеся эмулируют функции QOut() и QQOut(). Только один "набор" этих функций может быть задействован в одно и то же время. Это значит, что, например, не удастся вторично вызвать функцию RОutStart() без ROutEnd() для первого вызова. Тем не менее, если такое все-таки случится, первый вызов выведет текст на печать так, как если бы этот вызов был умышленно завершен.

Теперь нам нужен новый модуль в библиотеке Ch15 Library. Назовите его C:ROut Print. Откройте редактор кода и импортируйте код из файла Code4.prg с диска СD.

GLOBAL oROut AS _Rout

FUNCTION ROutStart(Owner, cFile)

Default(@cFile, NULL_STRING)

IF ! (oROut = NULL_OBJECT)

oROut:PrintString()

oROut := NULL_OBJECT

CollectForced()

ENDIF
oROut := _ROut{Owner, cFile}

RETURN NIL

FUNCTION ROut()

LOCAL dwCounter AS DWORD

LOCAL cStr AS STRING

FOR dwCounter :=1 UPTO PCount()

cStr += AsString(_GETFPARAM(dwCounter))

NEXT

oROut:AddString(cStr, TRUE)

RETURN NIL

FUNCTION RROut()

LOCAL dwCounter AS DWORD

LOCAL cStr AS STRING

FOR dwCounter :=1 UPTO PCount()

cStr += AsString(_GETFPARAM(dwCounter))

NEXT
oROut:AddString(cStr, FALSE)

RETURN NIL

FUNCTION ROutEnd()

oROut:PrintString()
IF SLen(Trim(oROut:cFile)) > 0

FClose(oROut:ptrHandle)

ENDIF
oROut:cText := NULL_STRING

oROut := NULL_OBJECT

CollectForced()

RETURN NIL

CLASS _ROut

PROTECT oOwner AS Window

EXPORT cText AS STRING

EXPORT cFile AS STRING

EXPORT ptrHandle AS PTR

METHOD Init(Owner, cFile) CLASS _ROut

SELF:oOwner := Owner

SELF:cText := NULL_STRING

SELF:cFile := cFile

IF SLen(Trim(cFile)) > 0

IF (SELF:ptrHandle := FCreate(cFile)) = F_ERROR

TextBox{ , "Ошибка создания выходного файла", "Ошибка создания файла: " ;

+ DosErrString(FError())}:Show()

SELF:cFile := NULL_STRING

BREAK

ENDIF

ENDIF

RETURN SELF
METHOD AddString(cString, lCRLF) CLASS _ROut

Default(@lCRLF, FALSE)

SELF:cText += cString

SELF:cText += iif(lCRLF, CRLF, NULL_STRING)

RETURN NIL

METHOD PrintString() CLASS _ROut

LOCAL oPrinter AS ReportMemo
IF SELF:cFile == NULL_STRING

oPrinter := ReportMemo{SELF:oOwner, "ROut Print", SELF:cText}

oPrinter:Start()

oPrinter:Destroy()

CollectForced()

ELSE

FWrite(SELF:ptrHandle, SELF:cText)

ENDIF

RETURN NIL
Прежде, чем мы продолжим, сохраните код. Функция ROutStart() создает объект класса _ROut, который сохраняется в глобальной переменной <oROut>, делая его доступным для всего приложения.

Функции ROut() и RROut() используют недокументированную функцию _GETFPARAM() для получения доступа к любому числу параметров любого типа, хотя наш код использует только те из них, которые могут иметь литеральное представление с помощью функции AsString(). ROut() добавляет символ перевода строки в конец строки значений, а RROut() - нет.

Использование функции _GETFPARAM() позволяет нашим функциям работать вообще без аргументов. Однако, использование недокументированных функций имеет и обратную сторону. В следующей версии этих функций может и не оказаться… Подобные полезные функции могут быть обнаружены при изучении CA-Visual Objects SDK.

Функция ROutEnd() печатает текст или выводит его в файл и закрывает его перед окончанием работы.

Основной класс _ROut сохраняет строку в виде свойства и обрабатывает опцию выбора сохранения ее в качестве строки или файла. Метод Init() открывает, если это требуется, файл и сохраняет указатель на него таким образом, чтобы он мог быть закрыт функцией ROutEnd().

Переменная <oOwner> необходима для вывода окна прерывания печати (Abort Dialog) во время печати.

Метод AddString() собирает различные части текста (строки) в единую строку, а метод PrintString() либо печатает упомянутую единую строку, либо сохраняет ее в файле. Если вывод перенаправлен в файл, с помощью ReportMemo эта же строка может быть напечатана, независимо от рассматриваемых сейчас функций.

Закройте редактор и перекомпилируйте библиотеку.

Пример использования ROut()

Этот пример довольно прост, но дает представление о легкости использования семейства функций ROut(). Для этого возвратитесь в модуль M:ShellMenu приложения Ch15 App и добавьте еще один пункт в меню Report - "List PrintingDevice Capabilities". Укажите имя события Event Name как "DevCaps". Сохраните изменения и закройте редактор.

Теперь в модуле C:ShellWindow откройте редактор кода и введите следующее:
METHOD DevCaps() CLASS ShellWindow

LOCAL uValue AS USUAL

BEGIN SEQUENCE

ROutStart(SELF)

uValue := oPrintDev:DeviceCapabilities(DC_PAPERNAMES)

ROut("Show " + "DC_PAPERNAMES")

PrintValue(uValue)

ROut()

ROut("Show " + "DC_PAPERS")

uValue := oPrintDev:DeviceCapabilities(DC_PAPERS)

PrintValue(uValue)

ROut()

ROutEnd()

END SEQUENCE

RETURN NIL

FUNCTION PrintValue(uValue)

IF IsArray(uValue)

AEval(uValue, {|u| PrintValue(u)})

ELSEIF IsNumeric(uValue) .AND. uValue = CAPABILITY_NOT_AVAILABLE

ROut ("Device capability not available from driver")

ELSE

ROut(uValue)

ENDIF

RETURN NIL
Сохраните код.

В строке 4 функции DevCaps() создается объект ROut(), а вызов ROutEnd() обеспечивает печать результирующей строки. ROut() вызывается как непосредственно, так и через PrintValue(), поэтому около 30 строковых объектов могут быть напечатаны одновременно в отдельных строках.

Вызов oPrintDec:DeviceCapabilities(DC_PAPERNAMES) возвращает массив, передаваемый далее в функцию PrintValue(), которая извлекает индивидуальные объекты из массива и передает каждый в ROut() для вывода на печать. Аналогично и для oPrintDev:DeviceCapabilities(DC_PAPERS).

Закройте редактор, скомпилируйте и запустите приложение. При выборе пункта меню List PrintingDevice Capabilities вы должны получить отчет примерно следующего вида:

Memo Print 

Show DC_PAPERNAMES

Letter

Legal

9

etc.
Если в 8-й строке функции DevCaps() заменить вызов ROut() на ROut(cEJECT), вместо пробельной строки мы получим разрыв страницы, поскольку ReportMemo трактует Chr(12) как команду выхода из цикла печати. (Ранее cEJECT определен как _Chr(12)). Однако, вместо ROut(cEJECT) лучше использовать RROut(cEJECT), поскольку он не добавляет символ перевода строки.

Как отмечалось ранее, при таком способе печати может быть использован только один шрифт. Тем не менее, в заголовке (Header) вы можете использовать иные шрифты, рисунки .bmp и т.д. и т.п.).

Ничто не мешает включению тэгов HTML в текст, выводимый с помощью ROut(). Все в ваших руках! Если вы создаете Web-страницы, то сможете найти и другие пути усовершенствования, хотя это уже не имеет прямого отношения к вопросам, рассматриваемым в настоящей главе.

Печать рисунков

Удобно поместить необходимый код для печати растровых изображений в класс ReportPrinter, поскольку требования к этим объектам все те же: печать определенного рисунка, в определенной области страницы, с определенным размером. Необходимо отметить, что в NT4 текущее разрешение экрана будет влиять на печать изображения. Перейдите в модуль C:ReportWriter в библиотеке Ch15 Library и введите следующий код:

METHOD PrintBitmap(aPoint, aDim, oBitmap) CLASS ReportPrinter

LOCAL oBitmapObject AS BitmapObject

LOCAL oDimension AS Dimension

LOCAL oPoint AS Point
IF ! SELF:NoPrint

oPoint := SELF:XYPoint(REAL4(aPoint[1]), REAL4(aPoint[2]))

oDimension := SELF:WHDimension(REAL4(aDim[1]), REAL4(aDim[2]))

oBitmapObject := BitmapObject{oPoint, oDimension, oBitmap}

SELF:Draw(oBitmapObject)

oBitmapObject:Destroy()

oDimension:Destroy()

CollectForced()

ENDIF

RETURN NIL
Сохраните код и давайте посмотрим, что же мы сделали. Мы получаем два массива, определяющих координаты Point и Dimension, а также сам рисунок в качестве параметров метода. В дальнейшем из этих параметров создаются объекты oPoint, oDimension и oBitmapObject. BitmapObject отличается от обычного Bitmap тем, что в нем уже определены его местоположение и размер. Затем он рисуется также, как мы делали это с ранее встречавшимися объектами DrawObject (LineObject и т.п.). В конце метода созданные объекты уничтожаются.

Закройте редактор и создайте новый в библиотеке Ch15 Library новый модуль с именем "C:BitMaps". откройте редактор кода и введите следующий код:

RESOURCE EVO_BMP Bitmap C:\EVO\Ch15\EVO.BMP
CLASS EVO_BMP INHERIT Bitmap
METHOD Init() CLASS EVO_BMP

SUPER:init(ResourceID{"EVO_BMP", _GetInst()})

RETURN SELF
Сохраните код, закройте редактор. Таким образом мы создали объект Bitmap и сохранили его в ресурсном примитиве, включаемом в дальнейшем в исполняемый файл. Необходимо, чтобы указанный файл рисунка существовал в месте, адресуемом введенным путем к нему. Это - код библиотеки. Вы можете заменить указанный файл своим собственным.

Теперь мы хотим добавить рисунок к нашему заголовку, используемому при печати. Перейдите в модуль C:ReportMemo, откройте в редакторе кода метод ReportMemoHeader:Print(). Измените код следующим образом:

METHOD Print() CLASS ReportMemoHeader

LOCAL oBitmap AS Bitmap 



// Новая строка
SELF:Owner:r4VerticalPosition := 99.0

SELF:Owner:PrintText(0, "Memo Print", ALIGN_LEFT)

SELF:Owner:PrintText(100, DToC(Today()), ALIGN_RIGHT)

oBitmap := EVO_BMP{} 




// Новая строка
SELF:Owner:PrintBitmap({35, 98}, {30, 4}, oBitmap) 
// Новая строка
oBitmap:Destroy() 




// Новая строка

SELF:Owner:PrintGrid({0,100}, {98.5, 98.5}, 2, "H", FALSE)

SELF:Owner:r4VerticalPosition := 0.0

SELF:Owner:PrintText(50, NTrim(SELF:Owner:dwPage), ALIGN_CENTRE)

SELF:Owner:r4VerticalPosition := 96.0

RETURN NIL
Теперь здесь создается объект Bitmap и он передается в метод PrintBitmap. Его координаты определены как 35, 98, а размер - 30, 4. Это означает, что рисунок будет растянут на 30% ширины листа и, чтобы центрировать его по горизонтали, необходимо вывести его в позиции (100-30)/2, или 35% от левого поля.

Сохраните код, закройте редактор и скомпилируйте библиотеку. Возвратитесь в приложение Ch15 App и запустите его. Теперь при печати мемо-отчета рисунок должен появится в области заголовка, если только вы не создавали подкласса ReportMemo и не определяли иного метода Print() (вы увидите это на диске CD).

Подсчет итогов в приложении Video
В этом разделе главы мы используем все три сервера из Ch15 Servers, те же самые, что мы использовали при создании приложения Video. В результате мы получим отчет, похожий на рис. 15.4. 

[image: image4.png]Puc. 154

OTYeT s NpUnoxeHIA Video ¢
MOAYETOM MPOMEKYTOUHBIX 1
oBluel cymm.

Loncud
R Objcets 2.5

Page Az Listing
Tile e Star Femdle s Cacgory  \ideo Fosiion  Fumring
Hhrber Time
Black Fain hifcheal Dougias retin ] 1
Darmolton b Syhestar Salne  Sanda Bullod 1 1 158
Dk Hard Bruce Wil H 1 145
D Hard 2 Bruce Wils H FREY
The Hunt For Rd Octcber  Sean Conrery 3 1
Top Gun Tam Cnize Kellphasiils a i 125
Sub-tral o Acion9AD
Podla 13 Tom Hanks rama s 1 —m
T Keu Costrer " 1D
Sub-tral fo: Drama__ 548
Fink Flowd n Cancert Pink Floyd Misic i (sl
‘The Seng Remans The Same  Lad Zeppein 7 1
Subtral fo: busig__538
ADay s The Racas hereBros. Comady e 1 ]
Wirang Git Hariscn Ford hilarie Gt 3 1 __1s

Sub-tral fo: Comady_207





Это самый сложный пример в данной главе. Он использует серверы для звезд женского и мужского полов. Он предотвращает повторную печать наименований категорий, запрещая печать, если значение в колонке не изменилось по сравнению с предыдущей записью. В завершение в отчете выводятся промежуточные результаты длительности воспроизведения при изменении категории фильма. Список женщин-звезд невелик.

Структура отчета может быть разбита на несколько частей для лучшего понимания процесса его создания. В отчете имеется заголовок, содержащий все, включая вторую длинную горизонтальную черту. Ниже расположены "тела" отчета, представляющие из себя строки. Каждый ряд или "тело" создается из соответствующей записи в Video2S и включает в себя поиск в серверах StarM и StarF и в массиве категорий. Промежуточные и общий итоги печатаются с помощью объекта Totals. Страница делится на колонки, используя класс ReportCol, хотя придется внести в него некоторые усовершенствования, чтобы сохранять значения итогов.

Мы возьмемся за дело поэтапно. Сначала мы напишем код в приложении, необходимый для вызова отчета. Затем создадим модуль в приложении и наследуемый класс Printer для отчета, но только для печати заголовка. Затем мы добавим тела отчета, а затем - суммы. Все это мы внесем в нашу библиотеку. На разных этапах мы будем проверять отсутствие ошибок.

В предыдущих отчетах мы добавляли очень немного кода в приложение для запуска библиотечного кода. В данном же случае придется поработать побольше.

Откройте ShellMenu в редакторе меню и добавьте в меню Reports еще один пункт с именем "Video Totals". Его свойству Event Name присвойте значение "VideoTotals". Сохраните меню и закройте редактор.

Перейдите в модуль C:ShellWindow, откройте редактор кода и введите код метода VideoTotals():

METHOD VideoTotals() CLASS ShellWindow

LOCAL oVideo AS Video2S

oVideo := Video2S{}

oVideo:SetOrder("Video2c")
BEGIN SEQUENCE

VideoPrint_TotalsReport(SELF, oVideo)

END
oVideo:Close()

RETURN NIL
Этот метод вызывает функцию, из которой запускается сам отчет, поскольку может оказаться удобным запускать генерацию отчета с помощью, например,  кнопки в окне данных. Функции могут быть вызваны откуда угодно и предоставляют нам свободу, которой грех не воспользоваться.

Сохраните код и закройте редактор. Теперь создайте новый модуль в приложении Ch15 App и назовите его "C:VideoTotals", откройте редактор кода и импортируйте файл Code5.prg c диска СD.

FUNCTION VideoPrint_TotalsReport(Owner, oServer)

LOCAL oPrinter AS VideoPrint

oPrinter := VideoPrint{Owner, "Video Print", oPrintDev, oServer}

oPrinter:Start()

oPrinter:Destroy()

CollectForced()

RETURN NIL

CLASS VideoPrint INHERIT ReportPrinter

PROTECT oServer AS DBServer

PROTECT oVP_Header AS VP_Header

PROTECT oCol1 AS ReportCol

PROTECT oCol2 AS ReportCol

PROTECT oCol3 AS ReportCol

PROTECT oCol4 AS ReportCol

PROTECT oCol5 AS ReportCol

PROTECT oCol6 AS ReportCol

PROTECT oCol7 AS ReportCol
METHOD Init(Owner,cJobName,oPrintingDevice, oServer) CLASS VideoPrint

LOCAL aTots AS ARRAY
SUPER:Init(Owner, cJobName, oPrintingDevice)

SELF:Owner := Owner

SELF:oServer := oServer

SELF:oServer:GoTop()

SELF:oVP_Header := VP_Header{SELF}

SELF:oCol1 := ReportCol{SELF, "", FALSE, 76, 6, TRUE, ALIGN_RIGHT}

SELF:oCol2 := ReportCol{SELF, "", FALSE, 83, 6, TRUE, ALIGN_RIGHT}

SELF:oCol3 := ReportCol{SELF, "", FALSE, 0, 35, TRUE, ALIGN_LEFT}

SELF:oCol6 := ReportCol{SELF, "", FALSE, 37,15, TRUE, ALIGN_LEFT}

SELF:oCol7 := ReportCol{SELF, "", FALSE, 52,13, TRUE, ALIGN_LEFT}

SELF:oCol4 := ReportCol{SELF, "", FALSE, 65, 8, FALSE, ALIGN_LEFT}

SELF:oCol5 := ReportCol{SELF, "", TRUE,  91, 8, TRUE, ALIGN_RIGHT}

SELF:Font := Font{FONTSWISS8}

SELF:r4VerticalLineHeight := 1.5

RETURN SELF
METHOD Destroy() CLASS VideoPrint

SELF:oVp_Header := NULL_OBJECT

SELF:oCol1 := NULL_OBJECT

SELF:oCol2 := NULL_OBJECT

SELF:oCol3 := NULL_OBJECT

SELF:oCol4 := NULL_OBJECT

SELF:oCol5 := NULL_OBJECT

SELF:oCol6 := NULL_OBJECT

SELF:oCol7 := NULL_OBJECT

SUPER:Destroy()

RETURN NIL
METHOD ReportExpose() CLASS VideoPrint

LOCAL lMorePages AS LOGIC

SELF:oVP_Header:Print()

RETURN lMorePages

ACCESS oCol1() CLASS VideoPrint

RETURN SELF:oCol1

ACCESS oCol2() CLASS VideoPrint

RETURN SELF:oCol2

ACCESS oCol3() CLASS VideoPrint

RETURN SELF:oCol3

ACCESS oCol4() CLASS VideoPrint

RETURN SELF:oCol4

ACCESS oCol5() CLASS VideoPrint

RETURN SELF:oCol5

ACCESS oCol6() CLASS VideoPrint

RETURN SELF:oCol6

ACCESS oCol7() CLASS VideoPrint

RETURN SELF:oCol7

ACCESS oServer CLASS VideoPrint

RETURN SELF:oServer

CLASS VP_Header

PROTECT Owner AS VideoPrint

METHOD Init(Owner) CLASS VP_Header

SELF:Owner := Owner

RETURN SELF

METHOD Print() CLASS VP_Header

LOCAL oBitmap AS Bitmap

LOCAL oDefaultFont AS Font
SELF:Owner:r4VerticalPosition := 97

SELF:PrintText(0, "Page" + NTrim(SELF:Owner:dwPage), ALIGN_LEFT)

oDefaultFont := SELF:Owner:Font

SELF:Owner:Font := Font{FONTSWISS14}

SELF:Owner:Font:Bold := TRUE

SELF:Owner:Font:Underline := TRUE

SELF:Owner:r4VerticalPosition := 97

SELF:Owner:PrintText(50, "Video Listing", ALIGN_CENTRE)

oBitmap := EVO_BMP{}

SELF:Owner:PrintBitmap({70, 96}, {30, 4}, oBitmap)

oBitmap:Destroy()

SELF:Owner:Font := oDefaultFont

SELF:Owner:r4VerticalPosition := 93

SELF:Owner:oCol1:Print("Video")

SELF:Owner:oCol2:Print("Position")

SELF:Owner:oCol3:Print("Title")

SELF:Owner:oCol4:Print("Category")

SELF:Owner:oCol5:Print("Running")

SELF:Owner:oCol6:Print("Male Star")

SELF:Owner:oCol7:Print("Female Star")

SELF:Owner:r4VerticalPosition -= SELF:Owner:r4VerticalLineHeight

SELF:Owner:oCol1:Print("Number")

SELF:Owner:oCol5:Print("Time")

SELF:Owner:PrintGrid({0, 100}, {0, 91, 100}, 2, "H", FALSE)

SELF:Owner:r4VerticalPosition := 89

CollectForced()

RETURN NIL
Сохраните код и давайте посмотрим, что мы сделали.

Прежде всего, вы можете спросить, зачем  <oServer> передается в качестве параметра в функцию VideoPrint_TotalsReport() из метода VideoTotals(), если можно обратиться к нему внутри самой функции. Однако, передача его в качестве параметра дает нам возможность установить фильтр, диапазон и т.п. прежде, чем передавать его в код генератора отчета. Ниже мы увидим, как можно сделать это.

В предыдущем отчете мы использовали массив объектов Column. В объявлении класса VideoPrint мы идентифицируем каждый из них индивидуально. Префикс VP используется в качестве аббревиатуры VideoPrint в свойствах и именах классов.  Класс ReportCol был разработан для печати отчета в виде телефонного списка, однако он также требует некоторых усовершенствований.

В методе Init() класса VideoPrint инициализируются различные свойства и самыми важными из них являются колонки. Они еще не находятся в нужном порядке. Их горизонтальные позиции определяются четвертыми параметрами, а следующий параметр определяет ширину каждой из них. Только <oCol4>, которая отображает категории, игнорирует дублирование, что определяется шестым параметром - FALSE. Размер ширины каждой из колонок имеет значение только когда они не выровнены по левому краю.

По миновании надобности вновь созданные объекты разрушаются в методе VideoPrint:Destroy(). Однако, SELF:Server здесь не закрывается, поскольку он был создан вне функции VideoPrint_ReportTotals() и будет закрыт там же. Также нет необходимости в дополнительном вызове CollectForced(), поскольку эта функция вызывается в методе ReportPrinter:Destroy().

В данном случае метод ReportExpose() сам по себе ничего не печатает. Ответственными за печать являются другие классы, которые в нашем случае ограничены одним - oVP_Header:Print().

Методы доступа по чтению Accesses необходимы для объектов ReportCol и сервера, соответственно, мы получаем доступ к ним из заголовка и других объектов.

В заголовке используются те же колонки, что и в объектах тела отчета и сумм для обеспечения правильности выравнивания текста.

Закройте редактор, скомпилируйте и запустите приложение. Выберите отчет Video Totals и запустите его для проверки правильности сделанного нами. На данном этапе вы должны получить одну страницу отчета, на которой будет напечатан только заголовок. Закройте приложение.

Следующей нашей задачей будет создание тел отчета, а подсчет итогов оставим напоследок.

Вернитесь в модуль C:VideoTotals, откройте объявление класса VideoPrint в редакторе кода и добавьте в него следующие свойства:

PROTECT oVP_Body AS VP_Body

PROTECT oMaleStar AS StarM

PROTECT oFemaleStar AS StarF
Сохраните класс, закройте редактор и откройте метод VideoPrint:Init(). Вставьте нижеследующие строки сразу после строки <SELF:oVP_Header>:

SELF:oVP_Header := VP_Header{SELF} 
// Существующая строка

SELF:oMaleStar := StarM{} 

// Новая строка
SELF:oFemaleStar := StarF{} 

// Новая строка
SELF:oVP_Body := VP_Body{SELF} 
// Новая строка
Закройте редактор и откройте метод ReportExpose(). Непосредственно перед возвратом из метода (RETURN) введите следующие строки:

DO WHILE SELF:r4VerticalPosition > 5 .AND. ! SELF:oServer:Eof

SELF:oVP_Body:Print()

SELF:oServer:Skip()

ENDDO

lMorePages := ! SELF:oServer:Eof
В этой части кода осуществляется проход по всем записям сервера и выполняется печать тела отчета (его еще предстоит разработать) вплоть до конца файла или до достижения конца страницы. Логическая переменная <lMorePages> сохраняет значение TRUE до достижения конца файла.

Сохраните код и закройте редактор.

Теперь нам нужны поисковые методы для поиска звезд - мужчин и женщин, которых мы впоследствии должны вставить в тело отчета. Откройте редактор кода и создайте следующие методы:

METHOD GetMaleStar(cMale) CLASS VideoPrint

LOCAL cRetVal := "" AS STRING

IF SELF:oMaleStar:Seek(cMale, FALSE)

cRetVal := SELF:oMaleStar:Names

ENDIF

RETURN cRetVal
METHOD GetFemaleStar(cFemale) CLASS VideoPrint

LOCAL cRetVal := "" AS STRING

IF SELF:oFemaleStar:Seek(cFemale, FALSE)

cRetVal := SELF:oFemaleStar:Names

ENDIF

RETURN cRetVal
Сохраните код и закройте редактор.

Объект Body ("тело" отчета) по окончании работы должен быть уничтожен. Кроме того, необходимо закрыть и уничтожить серверы по окончании печати. Откройте в редакторе кода метод VideoPrint:Destroy() и дополните его следующими строками непосредственно перед вызовом метода SUPER:Destroy():

SELF:oVP_Body := NULL_OBJECT

SELF:oMaleStar:Close()

SELF:oFemaleStar:Close()

SELF:oMaleStar := NULL_OBJECT

SELF:oFemaleStar := NULL_OBJECT

SUPER:Destroy() 

// Существующая строка

Сохраните код, закройте редактор.

Теперь нам нужен способ для интерпретации видео категории внутри тела отчета, чтобы в отчете было написано "Драма", а не "2", что соответствует ее значению. Для этого мы используем глобальный массив. Перейдите к модулю C:Start, создайте новый примитив и определите <aCategory> как глобальный массив.

GLOBAL aCategory AS ARRAY

Сохраните код, закройте редактор, а затем вновь откройте метод App:Start(). Добавьте две строки между вызовами методов Show() и Exec():

oShell:Show() 

// Существующая строка

aCategory := {} 

// Новая строка

FillCategory() 

// Новая строка

SELF:Exec() 

// Существующая строка
После добавления функции FillCategory() наименования категорий - "Драма", "Боевик" и т.п. будут доступны в приложении. Перейдите в конец метода Start(), пропустите одну строку и создайте функцию FillCategory():

FUNCTION FillCategory()

AAdd(aCategory, {"Action", "1"})

AAdd(aCategory, {"Drama", "2"})

AAdd(aCategory, {"Music", "3"})

AAdd(aCategory, {"Comedy", "4"})

RETURN NIL
Сохраните код, закройте редактор.

Нами не создан до сих пор класс тела отчета. Перейдите в модуль C:VideoTotals, откройте редактор кода и скопируйте в него следующее:

CLASS VP_Body

PROTECT Owner AS VideoPrint
METHOD Init(Owner) CLASS VP_Body

SELF:Owner := Owner

RETURN SELF
METHOD Print() CLASS VP_Body

LOCAL cCatCode, cCatString AS STRING

LOCAL dwPointer AS DWORD

cCatCode := SELF:Owner:oServer:Category

cCatString := ""

IF (dwPointer := AScan(aCategory, {|e| e[2] == cCatCode})) > 0

cCatString := aCategory[dwPointer, 1]

ENDIF
SELF:Owner:oCol1:Print(SELF:Owner:oServer:VNo)

SELF:Owner:oCol2:Print(SELF:Owner:oServer:Pos)

SELF:Owner:oCol3:Print(SELF:Owner:oServer:Title)

SELF:Owner:oCol4:Print(cCatString)

SELF:Owner:oCol5:Print(SELF:Owner:oServer:PlayTime)

SELF:Owner:oCol6:Print(SELF:Owner:GetMaleStar (SELF:Owner:oServer:StarM))

SELF:Owner:oCol7:Print(SELF:Owner:GetFemaleStar (SELF:Owner:oServer:StarF))

SELF:Owner:r4VerticalPosition -=SELF:Owner:r4VerticalLineHeight

RETURN NIL
Сохраните код. Метод Print() использует вызов AScan() для поиска и извлечения соответствующей категории для колонки 4. В колонки 5 и 6 передаются имена звезд, найденные методом класса VideoPrint GetMaleStar() и его "женским" эквивалентом. 

Закройте редактор, скомпилируйте и запустите приложение, чтобы удостовериться в корректности вывода тела отчета.

Нашей следующей задачей будет подробное изучение Итогов (Totals), что повлечет за собой необходимость модификации библиотеки Ch15 Library. Нам также предстоит разработать способ, как избежать повторения наименований категорий на странице.

Итоговые данные печатаются как отдельные объекты в момент изменения значения поля. Таким образом, например, для распечатки продаж в отдельно взятой стране файл сначала должен быть проиндексирован по полю кода страны, а затем отчет должен вывести итог в тот момент, когда код страны изменится. Нам нужно обеспечить печать итогов в момент перехода от одной записи к другой. Таким образом, мы выведем промежуточные итоги при каждой смене категории фильмов, а также общий итог в конце отчета.

В методе ReportExpose() цикл DO WHILE … ENDDO содержал вызов метода Print() для печати подробностей каждой записи. Теперь нам нужен еще метод LoadData() внутри которого следует выполнить сравнение и решить, следует ли выводить промежуточный итог. Он должен быть вызван до метода Print(). Фактически, мы разбиваем метод Print() на две части, а метод Total() будет вызываться между ними.

Концепция "уровней итогов" требует небольшого пояснения. Можно вывести итоговую сумму для всего отчета, хотя бы и для нескольких колонок. Если это единственные итоговые сведения в отчете, можно говорить о единственном уровне. Если же проиндексировать файл по полям Country+State+County, то можно создать промежуточные итоговые суммы для каждого блока записей, в которых значения Country+State+County одинаковы, еще один - где значения Country+State одинаковы, еще один - где одинаковы значения Country и общий итог по всему отчету. В этом случае можно говорить об отчете с четырьмя уровнями итоговых значений.

Мы используем эти преимущества для того, чтобы исключить печать значения в колонке, если предыдущее значение было таким же, при условии, что между ними нам не пришлось печатать заголовок или промежуточный итог.

Модификация библиотеки для обработки итогов

Метод LoadData() сравнивает текущее значение в колонке с предыдущим, соответственно, нам нужно добавить свойство для сохранения дополнительной информации. В классе уже существует переменная <cOldTextValue>, теперь нам понадобится другая переменная - <cNewTextValue>. 

Перейдите в модуль C:ReportPrinter библиотеки Ch15 Library, откройте объявление класса ReportCol. Добавьте в него следующие свойства:

INSTANCE cNewTextValue AS STRING

PROTECT lRepeat AS LOGIC

PROTECT aTotals AS ARRAY

PROTECT lNumeric AS LOGIC
Массив <aTotals> будет содержать значения промежуточных итогов для колонки, количество которых зависит от числа уровней итогов, заложенных в отчет. <lNumeric> определяет, должно ли значение в колонке иметь числовое представление.

Сохраните код, закройте редактор, и вновь откройте в нем метод ReportCol:Init(). В него необходимо добавить несколько строк, в которых переменные будут принимать значения, передаваемые в качестве уже существующих аргументов. Добавьте их непосредственно перед возвратом из метода:

SELF:lRepeat := lRepeat

SELF:aTotals := aTotals

SELF:lNumeric := lNumeric
Сохраните код, закройте редактор.

Нам необходимо убедиться в том, что при каждом изменении значения переменной <cNewTextValue> ее значение присваивается переменной <cOldTextValue>. Это очень хороший пример использования переменной типа INSTANCE, а не PROTECT. В случае переменной типа INSTANCE ее значение может быть изменено только через соответствующий метод ASSIGN, который и выполнит для нас всю работу. Создайте в редакторе новый примитив и введите следующий код:

ASSIGN cNewTextValue(cStr) CLASS ReportCol

SELF:cOldTextValue := SELF:cNewTextValue

SELF:cNewTextValue := Trim(cStr)

RETURN cStr
Помимо этого, нам нужен и соответствующий метод ACCESS для переменных <cNewTextValue> и <aTotals>, поэтому отступите на одну строку и введите код:

ACCESS cNewTextValue() CLASS ReportCol

RETURN SELF:cNewTextValue
ACCESS aTotals() CLASS ReportCol

RETURN SELF:aTotals
Метод LoadData() уже упоминался. Самое время написать его. Отступите еще на одну строку и введите код:

METHOD LoadData(cNewText, dwTotalLevel) CLASS ReportCol

LOCAL dwCounter AS DWORD
SELF:cNewTextValue := AllTrim(cNewText)
IF dwTotalLevel > 0 .AND. !SELF:cNewTextValue == SELF:cOldTextValue

IF ! SELF:Owner:lFirst

SELF:Owner:dwTotalsTrigger := dwTotalLevel

ENDIF

ENDIF
IF SELF:lNumeric

FOR dwCounter := 1 UPTO ALen(SELF:aTotals)

SELF:aTotals[dwCounter] += Val(SELF:cOldTextValue)

NEXT

ENDIF
RETURN NIL
Сохраните код и давайте посмотрим, что этот метод даст нам. В 3-й строке имеется вызов метода ASSIGN для <SELF:cNewTextValue>. Затем метод готовится к выполнению подсчета итогов, устанавливая <dwTotalsTrigger>, если это необходимо, в зависимость от изменения значения, за исключением случая обработки первой записи. Если колонка содержит численные данные, метод добавляет это значение во все элементы массива Totals. Впоследствии печать промежуточного итога обнулит распечатанный элемент массива. 

Закройте редактор и откройте метод ReportCol:Print(), который также требует модификации. Измените его следующим образом:

METHOD Print(uPrint) CLASS ReportCol

LOCAL cText AS STRING

LOCAL r4X, r4Y, r4Z AS REAL4

LOCAL oLine AS LineObject 


// Новая строка
r4X := SELF:r4XPos

r4Y := SELF:Owner:r4VerticalPosition

r4Z := SELF:Owner:r4VerticalLineHeight

IF UsualType(uPrint) = VOID

cText := SELF:cNewTextValue 

// Строка изменена
ELSEIF ValType(uPrint) == "C"

cText := uPrint

ELSEIF ValType(uPrint) == "N" 


// Начало новой секции

cText := AsString(SELF:aTotals[uPrint])

SELF:Owner:lCancelRepeatBars := TRUE

oLine := LineObject{SELF:Owner:XYPoint(r4X, r4Y), SELF:Owner:XYPoint(r4X + SELF:r4Width, r4Y)}

SELF:Owner:Draw(oLine)

IF uPrint = 1

oLine := LineObject{SELF:Owner:XYPoint(r4X, r4Y+ r4Z), SELF:Owner:XYPoint(r4X ;

+ SELF:r4Width, r4Y + r4Z)}

ENDIF

SELF:Owner:Draw(oLine)

ENDIF
IF (!SELF:cNewTextValue == SELF:cOldTextValue) .OR. SELF:lRepeat .OR. SELF:Owner:lCancelRepeatBars

// Конец новой секции
IF SELF:dwALIGN = ALIGN_RIGHT

r4X += SELF:r4Width

ENDIF

IF SELF:dwALIGN = ALIGN_CENTRE

r4X += SELF:r4Width / 2

ENDIF

SELF:Owner:PrintText(r4X, cText, SELF:dwAlign)

ENDIF

RETURN NIL
Сохраните код, закройте редактор. <oLine> используется для отчеркивания итогового значения. Такое происходит в случае, когда числовое значение передается методу в переменной <uPrint>, а это случается только когда этот метод вызывается из секции Totals отчета. Это значение не выводится на печать. На печать выводится значение из <SELF:aTotals>, определяемое значением TotalLevel.

Конструкция IF в конце новой секции определяет, должен ли данный объект печати быть напечатан в колонке. Он печатается, если его значение отличается от предыдущего, если установлен признак повторной печати в колонке или переменная <lCancelRepeatBars> владельца установлена в значение TRUE. Итоговые значения будут печататься всегда, потому что значение TRUE установлено в самом методе.

Теперь нам нужно модифицировать ReportPrinter для того, чтобы он работал с ReportCol. Сначала добавьте некоторые свойства в объявление класса ReportPrinter перед объявлением типизированных методов:

PROTECT lCancelRepeatBars AS LOGIC

PROTECT lFirst AS LOGIC

PROTECT dwTotalsTrigger AS DWORD
Сохраните код, закройте редактор и откройте метод ReportPrinter:Init(). В конце метода вы увидите вызов методов Destroy() для объектов <o_Origin> и <oDimension>. Сразу после этих строк добавьте:

oDimension:Destroy() 

// Существующая строка
SELF:lFirst := TRUE 

// Новая строка
SELF:lCancelRepeatBars := TRUE 
// Новая строка
Сохраните код, закройте редактор.

Создайте новый примитив и скопируйте в него следующие стандартные (т.е. никаких проверок и т.п.) методы ACCESS и ASSIGN:

ACCESS lCancelRepeatBars() CLASS ReportPrinter

RETURN SELF:lCancelRepeatBars
ASSIGN lCancelRepeatBars(L) CLASS ReportPrinter

SELF:lCancelRepeatBars := L

RETURN L
ACCESS lFirst() CLASS ReportPrinter

RETURN SELF:lFirst
ASSIGN lFirst(L) CLASS ReportPrinter

SELF:lFirst := L

RETURN L
ACCESS dwTotalsTrigger() CLASS ReportPrinter

RETURN SELF:dwTotalsTrigger

ASSIGN dwTotalsTrigger(n) CLASS ReportPrinter

SELF:dwTotalsTrigger := n

RETURN n
Сохраните код, закройте редактор. Перекомпилируйте библиотеку, удостоверьтесь, чем весь код введен правильно - компиляция должна пройти без ошибок и предупреждений. Тем не менее, мы не сможем проверить библиотеку, пока не внесем необходимые изменения в код нашего приложения. Это наша следующая задача.

Итоги

Прежде, чем приступить к печати, нам нужно разбить метод VP_Body:Print() на две части - метод LoadData() и укороченный метод Print().

Перейдите в приложение Ch15 App, выберите модуль C:VideoTotals и откройте метод VP_Body:Print() в редакторе кода. Выделите весь код метода, скопируйте его в буфер обмена комбинацией клавиш <Ctrl+C>, перейдите в конец метода, отступите строку-другую и вставьте содержимое буфера с помощью комбинации клавиш <Ctrl+V>. Измените первую строку одной из копий на -

METHOD LoadData() CLASS VP_Body

Теперь модифицируйте этот метод следующим образом:

METHOD LoadData() CLASS VP_Body

LOCAL cCatCode, cCatString AS STRING

LOCAL dwPointer AS DWORD
cCatCode := SELF:Owner:oServer:Category

cCatString := ""
IF (dwPointer := AScan(aCategory, {|e| e[2] == cCatCode})) > 0

cCatString := aCategory[dwPointer, 1]

ENDIF

// Здесь начинаются изменения…

SELF:Owner:oCol1:LoadData(SELF:Owner:oServer:VNo, 0)

SELF:Owner:oCol2:LoadData(SELF:Owner:oServer:Pos, 0)

SELF:Owner:oCol3:LoadData(SELF:Owner:oServer:Title, 0)

SELF:Owner:oCol4:LoadData(cCatString, 1)

SELF:Owner:oCol5:LoadData(SELF:Owner:oServer:PlayTime, 0)

SELF:Owner:oCol6:LoadData(SELF:Owner:oServer:StarM, 0)

SELF:Owner:oCol7:LoadData(SELF:Owner:oServer:StarF, 0)

SELF:Owner:lFirst := FALSE

RETURN NIL
Метод VP_Body:LoadData() подготавливает объект Column путем поиска того, что должно быть напечатано далее. Второй параметр, передаваемый в метод ReportCol:LoadData(), является числовым. Если этот параметр равен нулю, значит эта колонка не играет никакой роли в подсчете итоговых значений. Для четвертой колонки этот параметр устанавливается в значение единицы. Это означает, что итоговое значение первого уровня должно быть напечатано при изменении значения этой колонки. Последней задачей этого метода является изменение значения <lFirst> на FALSE, поскольку первая запись обрабатывается несколько иначе, нежели другие.

Метод ReportCol:LoadData(), вызываемый из метода VP_Body:LoadData(), устанавливает значение <SELF:dwTotalsTrigger>. Если значение этой переменной больше 0, то одна или несколько линий печатаются перед выводом тела отчета. Возвращаясь к аналогии Country+State+County, при изменении значения страны (Country) необходимо вывести промежуточный итог для последнего графства (County) этого штата (State), а сразу после него выводится промежуточный итог для штата.

В другой копии бывшего метода VP_Body:Print() необходимо лишь удалить несколько строк текста. Удалите все строки от первого объявления переменной LOCAL до конца цикла - ENDIF, а для колонок 1-5 метод LoadData() сделал уже все необходимое. Следовательно, нужно только удалить текст внутри скобок. Окончательно метод должен выглядеть так:

METHOD Print() CLASS VP_Body

SELF:Owner:oCol1:Print()

SELF:Owner:oCol2:Print()

SELF:Owner:oCol3:Print()

SELF:Owner:oCol4:Print()

SELF:Owner:oCol5:Print()

SELF:Owner:oCol6:Print(SELF:Owner:GetMaleStar( SELF:Owner:oServer:StarM))

SELF:Owner:oCol7:Print(SELF:Owner:GetFemaleStar( SELF:Owner:oServer:StarF))

SELF:Owner:r4VerticalPosition -= SELF:Owner:r4VerticalLineHeight

RETURN NIL
Сохраните код, закройте редактор, и откройте метод VideoPrint:ReportExpose(). Вставьте строку

SELF:oVP_Body:LoadData()
выше строки

SELF:oVP_Body:Print()

Сохраните код, закройте редактор, скомпилируйте и запустите приложение. Отчет Video Totals должен напечататься как и ранее, без итогов.

Откройте объявление класса VideoPrint и добавьте в него еще одно свойство.

PROTECT oVP_Totals AS VP_Totals

Сохраните код и перейдите в метод VideoPrint:Init(), где нем нужно обеспечить присвоение значения вновь введенной переменной. Добавьте нижеследующую строку после аналогичного присвоения значения переменной <oVP_Body>.

SELF:oVP_Body := VP_Body{SELF} 
// Существующая строка

SELF:oVP_Totals := VP_Totals{SELF} 
// Новая строка
В том же самом методе замените строку, в которой присваивалось значение переменной <SELF:oCol5> тремя новыми строками:

SELF:oCol4 := ReportCol{SELF, "", FALSE, 65, 8, FALSE, ALIGN_LEFT} 
 // Существующая строка
aTots := ArrayNew(2) 






 // Новая строка
AFill(aTots,0) 







 // Новая строка
SELF:oCol5 := ReportCol{SELF, "", TRUE, 91, 8, TRUE, ALIGN_RIGHT, aTots} // Новая строка
SELF:Font := Font{FONTSWISS8} // Existing Line
Массив <aTots> содержит 2 элемента, поскольку в нашем отчете существуют 2 уровня подведения итогов: первый - при смене Категории, а второй - общий итог. Итог подводится только по этой колонке. Сохраните изменения и закройте редактор.

Объект Totals должен быть уничтожен, соответственно, перейдите в метод VideoPrint:Destroy() и добавьте строку:

SELF:oVP_Totals := NULL_OBJECT 
// Новая строка
SUPER:Destroy() 


// Существующая строка
Сохраните метод и закройте редактор.

Теперь нам нужно снова изменить метод VideoPrint:ReportExpose(). Новый метод целиком с внесенными изменениями приведен ниже:

ReportExpose() CLASS VideoPrint

LOCAL lMorePages AS LOGIC

SELF:oVP_Header:Print()

DO WHILE SELF:r4VerticalPosition > 5 .AND. ! SELF:oServer:Eof

SELF:oVP_Body:LoadData() 

// Перемещено
IF SELF:dwTotalsTrigger > 0 

// Новая строка
SELF:oVP_Totals:Print(FALSE) 
// Новая строка
ENDIF 




// Новая строка

SELF:oVP_Body:Print()

SELF:oServer:Skip()

ENDDO

SELF:oVP_Totals:Print(TRUE) 


// Новая строка
lMorePages := ! SELF:oServer:Eof

RETURN lMorePages
Приведенный здесь метод вызывает метод oVP_LoadData() и, если последний изменяет значение переменной <SELF:dwTotalsTrigger>, печатается итоговое значение. По окончании цикла ENDDO oVP_Totals вызывается вновь, поскольку это необходимо по окончании печати последнего из тел отчета. В этот момент в метод передается значение TRUE, вследствие чего он "понимает", что имеет дело с последней записью.

Сохраните измененный метод и закройте редактор.

Наконец, нам нужен и сам класс VP_Totals. Откройте C:VideoTotals в редакторе кода и добавьте в него три примитива:

CLASS VP_Totals

PROTECT Owner AS VideoPrint
METHOD Init(oOwner) CLASS VP_Totals

SELF:Owner:= oOwner

RETURN SELF
METHOD Print(lEndReport) CLASS VP_Totals

LOCAL dwTotalLevel, dwCounter AS DWORD

IF lEndReport

SELF:Owner:dwTotalsTrigger := 2

FOR dwCounter := 1 UPTO ALen(SELF:Owner:oCol5: aTotals)

SELF:Owner:oCol5:aTotals[dwCounter] += Val(SELF:Owner:oCol5:cNewTextValue)

NEXT

ENDIF
FOR dwTotalLevel := 1 UPTO SELF:Owner:dwTotalsTrigger

IF dwTotalLevel = 1

SELF:Owner:PrintText(78, "Sub-total for: " + SELF:Owner:oCol4:cOldTextValue, ALIGN_LEFT)

ELSE

SELF:Owner:PrintText(85, "Grand Total", ALIGN_LEFT)

ENDIF

SELF:Owner:oCol5:Print(dwTotalLevel)

SELF:Owner:oCol5:aTotals[dwTotalLevel] := 0

SELF:Owner:r4VerticalPosition -= SELF:Owner:r4VerticalLineHeight * 1.5

NEXT
SELF:Owner:dwTotalsTrigger := 0

RETURN NIL
Переменная <lCancelRepeatBars> устанавливается в значение TRUE в методе ReportCol:Print() при печати итоговых значения для того, чтобы удостовериться в том, что значения напечатаны и для подготовки к обработке следующего тела отчета. Она также принимает значение TRUE в конце работы методов заголовка (Header) и значение FALSE в конце метода Print() класса Body. Это сделано для того, чтобы проконтролировать, что колонки, которые должны следовать правилу о повторности, выполнят его и выведут соответствующее значение при следующем вызове метода Print() класса Body. В нашем случае это имеет ограниченное применение, поскольку для решения вопроса - выводить или не выводить итоговые данные на печать - используется одна колонка. В любом случае, перед вызовом функции CollectForced в конце метода VP_Header:Print() необходимо добавить строку:

SELF:Owner:lCancelRepeatBars := TRUE

и

SELF:Owner:lCancelRepeatBars := FALSE
перед возвратом RETURN из метода VP_Body:Print().

Если заглянуть в метод VP_Totals:Print(), то мы увидим, что при окончании отчета переменная <dwTotalsTrigger> принимает значение "2", которое выше, чем значение числа промежуточного уровня отчета, соответственно, в массив добавляются значения для последней записи. Затем печатается строка примерно такого содержания: "Sub-total for: Drama" ("промежуточный итог для: Драма"). В конце печатается общее итоговое значение в колонке 5, после чего это значение обнуляется, а <dwTotalsTrigger> также принимает значение 0.

Сохраните код, закройте редактор, скомпилируйте и запустите приложение. Теперь вы можете напечатать весь отчет полностью, с промежуточными и окончательным итогом.

Код для данного отчета не был тривиальным, однако весьма значительную его часть можно использовать от отчета к отчету. Билиотека, конечно же, может быть улучшена, что потребует написания меньшего кода непосредственно в приложениях, однако объяснить это сейчас было бы затруднительно. Теперь становится понятнее популярность визуальных редакторов, однако все написанное выше целиком и полностью находится под вашим контролем и во многих случаях обладает несомненными преимуществами.

Если вы добрались до этого места, копируя и вводя код, тогда примите наши поздравления. Возможно, вам станет легче, если вы узнаете, что более никакого кода вручную вводить не придется.

Печать отчета Video Totals с помощью PushButton

Приложение ReportPrinter.aef на компакт-диске содержит такой элемент меню как Video Browser, который позволяет просмотреть данные из файла Vide2S. Этот код был скопирован из главы 11. Импортируйте три экспортных файла и библиотеку с диска и скомпилируйте библиотеку и серверы прежде, чем начнете компилировать приложение.

Код на CD имеет строчные комментарии в отличие от вашей "ручной" версии.

После импорта и компиляции запустите приложение и выберите VideoBrowser. В этой версии имеется кнопка PRINT, которая вызовет отчет Video Totals, который мы разработали выше. В дополнение к этому имеется еще кнопка SEARCH, которая выводит окно поиска, позволяющее ограничить область поиска заданной категорией. Если в броузере установлен фильтр с указанием какой-либо конкретной категории, то нажатие на кнопку PRINT приведет к созданию отчета только по указанной категории.

Использование функции для запуска генератора отчета делает очень простым добавление кнопки PRINT в окно броузера и использование всех его возможностей.

Вот необходимый код:

METHOD pbPrint() CLASS BrowseWin

LOCAL dwRecno AS DWORD

LOCAL cIndex AS STRING
dwRecno := SELF:SubServer:Recno

cIndex := SELF:SubServer:OrderInfo(DBOI_INDEXNAME)

SELF:SubServer:SuspendNotification()

SELF:SubServer:SetOrder("Video2c")

SELF:oSFSubBrowseWin:Browser:SuspendUpdate()

VideoPrint_TotalsReport(SELF, SELF:SubServer)

SELF:oSFSubBrowseWin:Browser:RestoreUpdate()

SELF:SubServer:ResetNotification()

SELF:SubServer:Recno := dwRecno

SELF:SubServer:SetOrder(cIndex)

SELF:SubServer:Notify(NOTIFYFILECHANGE)

RETURN NIL
В приведенном коде строка

VideoPrint_TotalsReport(SELF, SELF:SubServer)

печатает отчет, передавая в окно AbortDialog SELF в качестве владельца, а также сервер, который может быть (а может и не быть) отфильтрован, в зависимости от выбора пользователя. Остальной код в pbPrint() устанавливает новый индекс, а затем восстанавливает старый, обеспечивая неизменность положения указателя записи и их последовательность.

Печать файла memo
Еще одной особенностью, которую вы обнаружите после импорта файлов, является возможность распечатки мемо-файла. Наша глава и так достаточно велика без того, чтобы вдаваться в детали. Вы увидите, что эта опция открывает стандартное диалоговое окно выбора файла, по умолчанию предлагается выбрать текстовый файл *.TXT в корневом каталоге диска С, хотя вы, конечно, можете изменить маску на *.INI или любые другие форматы текстовых файлов и в любом каталоге. После указания имени файла выводится окно DataDialog. В нем  вы можете указать число требуемых копий, диапазон печатаемых страниц, например, со 2-й по 4-ю, а также ориентацию листа бумаги. Тип используемого окна DataDialog, поскольку оно модально и содержит FieldSpecs. Метод Window:PreValidate() был использован для включения/выключения однострочных элементов ввода From ("с") и To ("по"), поскольку этот метод вызывается всякий раз, когда фокус ввода перемещается из одного элемента к другому.

Для вывода используется класс Ex_Memo, который унаследован от класса ReportMemo, единственное изменение в нем заключается в выводе имени файла при печати заголовка, а рисунок, вставлявшийся по умолчанию, на печать не выводится. Версия класса ReportMemo, импортированная с CD, отличается от разработанной нами только тем, что она обеспечивает последовательное чтение данных блоками по 5 Кб каждый.

Вы можете также обратить внимание на способ использования окна OpenDialog в методе PrintFile() класса ShellWindow, а также можете модифицировать метод PrintString(), разработанный выше таким образом, чтобы устранить жесткое кодирование строки с именем файла. 

Заключение

Эта глава была очень длинной, со множеством кода. Возможно, для некоторых слишком много кода и слишком мало картинок! Глава преследовала по меньшей мере две цели: создать класс Printer и предоставить примеры объектно-ориентированного кода. В частности, мы увидели преимущества использования наследования и гибкость языка.

Надеемся, что глава поможет начинающим почувствовать, что они могут создавать отчеты без знания ODBC и без использования каких-либо других программных пакетов. Недостатки класса очевидны всякому, кто использовал альтернативные генераторы отчетов, однако, как бы то ни было, класс требует к себе пристального внимания и многие люди предпочитают пользоваться именно им.

Были, разумеется, и сложности, которые нам удалось преодолеть. Мы разобрались, как преобразовать страничные координаты в проценты, как устранить небольшие различия в аппаратно определенных полях и как обеспечить печать заданного диапазона страниц. Сделав это, мы получили несколько вариантов отчетов.

Нет никаких препятствий для использования класса Printer и какого-либо внешнего генератора отчетов в приложении одновременно, если в этом есть необходимость.

Втайне мы надеемся, что вы найдете способы для улучшения представленной здесь библиотеки, но рассмотрение самых желательных вариантов сделало бы изложение слишком запутанным. Даже без изменения имеющихся библиотечных модулей можно добавить модуль C:ReportLabels.

Теперь вы понимаете, почему мы назвали класс Printer Золушкой CA-Visual Objects?

